```
> # Construction of strictly positive polynomials in the boundary of the SOS cone with unique SOS decomposition.
  # See the details in Section 5 of "Strictly positive polynomials in the border of the SOS cone", by S. Laplagne and M. Valdettaro.
> ##########################################################################
  ## Section 5.2
  ## Polynomials of 5 variables in degree 4
  ##########################################################################
> ##########################################################################
  # Load "Rational SOS" procedures
  ##########################################################################
  read("rationalSOS.mpl");
  with(rationalSOS);
  with(LinearAlgebra);
```

"Opening connection with Matlab"

rationalSOS := **module**( ) ... **end module**

[*cancelDenominator*, *decompositionToMatrix*, *dimSimplex*, *evalMat*, *evalSolution*, *exactSOS*, *getCoeffs*, *getDiag*, *getExtension*, *getVars*, *homogenize*, *isHomogeneous*, *linIndepRows*, *listSubsets*, *matrixToPoly*, *minorsDet*, *nonRatCoef*, *numericSolver*, *numericSolverSubmatrix*, *numericSolverSubmatrixMaxRank*, *numericSolverSubmatrixRoundBefore*, *polyToMatrix*, *polyToMatrixVars*, *primitiveMatrix*, *randomRank*, *reduceByLinearEquation*, *reduceByLinearEquationLinear*, *roundAbs*, *roundMat*, *roundMatFloat*, *roundMatToZero*, *roundToIntMatrix*, *roundVec*, *rround*, *rrounde*, *sedumiCallMaxSpectralNorm*, *sedumiCallMaxSpectralNormSDP*, *sedumiCallObjective*, *smallToZero*, *smallToZeroMatrix*, *solveSubmatrixGeneral*, *solveToZero*, *vectorTrace*, *zeroDetSRows*, *zeroDetSys*, *zeroRows*]

[*&x*, *Add*, *Adjoint*, *BackwardSubstitute*, *BandMatrix*, *Basis*, *BezoutMatrix*, *BidiagonalForm*, *BilinearForm*, *CARE*, *CharacteristicMatrix*, *CharacteristicPolynomial*, *Column*, *ColumnDimension*, *ColumnOperation*, *ColumnSpace*, *CompanionMatrix*, *CompressedSparseForm*, *ConditionNumber*, *ConstantMatrix*, *ConstantVector*, *Copy*, *CreatePermutation*, *CrossProduct*, *DARE*, *DeleteColumn*, *DeleteRow*, *Determinant*, *Diagonal*, *DiagonalMatrix*, *Dimension*, *Dimensions*, *DotProduct*, *EigenConditionNumbers*, *Eigenvalues*, *Eigenvectors*, *Equal*, *ForwardSubstitute*, *FrobeniusForm*, *FromCompressedSparseForm*, *FromSplitForm*, *GaussianElimination*, *GenerateEquations*, *GenerateMatrix*, *Generic*, *GetResultDataType*, *GetResultShape*, *GivensRotationMatrix*, *GramSchmidt*, *HankelMatrix*, *HermiteForm*, *HermitianTranspose*, *HessenbergForm*, *HilbertMatrix*, *HouseholderMatrix*, *IdentityMatrix*, *IntersectionBasis*, *IsDefinite*, *IsOrthogonal*, *IsSimilar*, *IsUnitary*, *JordanBlockMatrix*, *JordanForm*, *KroneckerProduct*, *LA_Main*, *LUDecomposition*, *LeastSquares*, *LinearSolve*, *LyapunovSolve*, *Map*, *Map2*, *MatrixAdd*, *MatrixExponential*, *MatrixFunction*, *MatrixInverse*, *MatrixMatrixMultiply*, *MatrixNorm*, *MatrixPower*, *MatrixScalarMultiply*, *MatrixVectorMultiply*, *MinimalPolynomial*, *Minor*, *Modular*, *Multiply*, *NoUserValue*, *Norm*, *Normalize*, *NullSpace*, *OuterProductMatrix*, *Permanent*, *Pivot*, *PopovForm*, *ProjectionMatrix*, **(1)** *QRDecomposition*, *RandomMatrix*, *RandomVector*, *Rank*, *RationalCanonicalForm*, *ReducedRowEchelonForm*, *Row*, *RowDimension*, *RowOperation*, *RowSpace*, *ScalarMatrix*, *ScalarMultiply*, *ScalarVector*, *SchurForm*, *SingularValues*, *SmithForm*, *SplitForm*, *StronglyConnectedBlocks*, *SubMatrix*, *SubVector*, *SumBasis*, *SylvesterMatrix*, *SylvesterSolve*, *ToeplitzMatrix*, *Trace*, *Transpose*, *TridiagonalForm*, *UnitVector*, *VandermondeMatrix*, *VectorAdd*, *VectorAngle*, *VectorMatrixMultiply*, *VectorNorm*, *VectorScalarMultiply*, *ZeroMatrix*, *ZeroVector*, *Zip*]

```
> # Display tables of any size
interface(rtablesize = infinity);
```

$$10 \tag{2}$$

```
> ###########################################################################
## Example 5.5
## Example in the border with unique solution
###########################################################################
```

```
> # The first 4 polynomials correspond to an example of a polynomial
# in the non-negative border of the SOS-cone in the 4-4 case.
# We add a fifth polynomial to produce an example for the 5-4 case.
p1 := x1^2 - x4^2;
p2 := x2^2 - x4^2;
p3 := x3^2 - x4^2;
p4 := -x1^2 - x1 * x2 - x1 * x3 + x1 * x4 - x2 * x3 + x2 * x4 + x3 * x4;
p5 := x5^2;
```

$$p1 := x1^2 - x4^2$$
$$p2 := x2^2 - x4^2$$
$$p3 := x3^2 - x4^2$$
$$p4 := -x1^2 - x1\,x2 - x1\,x3 + x1\,x4 - x2\,x3 + x2\,x4 + x3\,x4$$
$$p5 := x5^2 \tag{3}$$

```
> # f is the sum of squares of p1, ..., p5
f := p1^2 + p2^2 + p3^2 + p4^2 + p5^2;
```

$$f := (x1^2 - x4^2)^2 + (x2^2 - x4^2)^2 + (x3^2 - x4^2)^2 + (-x1^2 - x1\,x2 - x1\,x3 + x1\,x4 - x2\,x3 + x2\,x4 + x3\,x4)^2 + x5^4 \tag{4}$$

```
> # We use SEDUMI to compute a SOS decomposition.
# With default options, exactSOS will compute a solution of maximum rank
out := exactSOS(f, facial = "no") :
```

"Number of indeterminates: ", 50

"Calling numerical solver SEDUMI to find the values of the remaining indeterminates..."

"SEDUMI CALL - eig"

"An exact positive definite solution could not be found for the reduced problem."

"Computing Cholesky decomposition..."
"Cholesky decomposition failed. Matrix is not positive definite. We use LDLt decomposition."
"Matrix decomposition failed for output matrix. Please check!" **(5)**

```
> # out[3] is a matrix in the spectrahedron of maximum rank.
  # We check the eigenvalues to determine the rank
  eig(out[3]);
```

$$\begin{bmatrix} -5.90279670642837\ 10^{-16} \\ -1.46027594269019\ 10^{-17} \\ -1.72450464225737\ 10^{-19} \\ 5.28963492790034\ 10^{-27} \\ 6.97511604460007\ 10^{-27} \\ 1.45569638669311\ 10^{-16} \\ 3.32386678373941\ 10^{-16} \\ 1.44466860600000\ 10^{-10} \\ 1.56692265300000\ 10^{-10} \\ 1.60974420482516\ 10^{-10} \\ 0.888960947926901 \\ 0.999999999948923 \\ 1.00000000005108 \\ 3.89989969879447 \\ 7.21113935327862 \end{bmatrix} \quad \textbf{(6)}$$

```
> # There are only 5 non-zero eigenvalues, the maximum rank in the
   # spectrahedron is 5.
```

```
> # We compare the matrix obtained by SEDUMI with the matrix corresponding
  # to the original decomposition p1^2+p2^2+p3^2+p4^2+p5^2.
  v := convert(out[5], list) : # The monomials indexing the columns of the Gram Matrix
  A1 := decompositionToMatrix([p1, p2, p3, p4, p5], v) :
  A2 := out[3] :
   evalf(Norm(A1 - A2));
```

$$2.310669740\ 10^{-10} \quad \textbf{(7)}$$

```
> # We see that both matrices are the same.
  # This gives strong numerical evidence that this is the unique matrix
  # in the spectrahedron of f.
```

```
> ###########################################################################
  ## Example 5.6
  ## Example in the border with a matrix in the spectrahedron of rank 6.
  ###########################################################################
```

```
> # We add a polynomial p6 to example 5.5
p1 := x1^2- x4^2;
p2 := x2^2- x4^2;
p3 := x3^2- x4^2;
p4 := -x1^2 - x1*x2 - x1*x3 + x1*x4 - x2*x3 + x2*x4 + x3*x4;
p5 := x5^2;
p6 := x1*x5 + x4*x5;
f := p1^2 + p2^2 + p3^2 + p4^2 + p5^2 + p6^2;
```

$$p1 := x1^2 - x4^2$$

$$p2 := x2^2 - x4^2$$

$$p3 := x3^2 - x4^2$$

$$p4 := -x1^2 - x1\,x2 - x1\,x3 + x1\,x4 - x2\,x3 + x2\,x4 + x3\,x4$$

$$p5 := x5^2$$

$$p6 := x1\,x5 + x4\,x5$$

$$f := (x1^2 - x4^2)^2 + (x2^2 - x4^2)^2 + (x3^2 - x4^2)^2 + (-x1^2 - x1\,x2 - x1\,x3 + x1\,x4 - x2\,x3 + x2\,x4 + x3\,x4)^2 + x5^4 + (x1\,x5 + x4\,x5)^2 \quad \textbf{(8)}$$

```
> # We use SEDUMI to compute a SOS decomposition.
# With default options, exactSOS will compute a solution of maximum rank
> out := exactSOS( f, facial = "no", computePolynomialDecomposition = "no") :
```

"Number of indeterminates: ", 50

"Calling numerical solver SEDUMI to find the values of the remaining indeterminates..."

"SEDUMI CALL - eig" **(9)**

```
> # out[3] is a matrix in the spectrahedron of maximum rank.
# We check the eigenvalues to determine the rank
eig(out[3]);
```

(10)

$$\begin{bmatrix} -6.66230748926793\ 10^{-16} \\ -2.75657741605117\ 10^{-16} \\ -1.41250904958926\ 10^{-16} \\ 4.81989475941244\ 10^{-17} \\ 1.15352713077803\ 10^{-16} \\ 3.31852698856289\ 10^{-16} \\ 1.07280866950797\ 10^{-8} \\ 2.37835979991430\ 10^{-8} \\ 7.85003461025040\ 10^{-8} \\ 0.888960947926640 \\ 0.999999999999999 \\ 1.00000000000025 \\ 1.99999995690006 \\ 3.89989969879448 \\ 7.21113935327863 \end{bmatrix} \quad (10)$$

```
> # There are 6 non-zero eigenvalues.
> # We compare the matrix obtained by SEDUMI with the matrix corresponding
  # to the original decomposition p1^2+p2^2+p3^2+p4^2+p5^2+p6^2.
  v := convert(out[5], list) : # The monomials indexing the columns of the Gram Matrix
  A1 := decompositionToMatrix([p1, p2, p3, p4, p5, p6], v) :
  A2 := roundToIntMatrix(out[3], 6) : # We convert some almost integer values to integers
   evalf(Norm(A1 - A2));
```

$$0. \quad (11)$$

```
> # We see that both matrices are the same.
  # This gives strong numerical evidence that this is the unique matrix
  # in the spectrahedron of f.
> ################################################################################
  ## Example 5.7
  ## Sum of 7 squares with unique decomposition
  ################################################################################

> # The first 4 polynomials correspond to an example of a polynomial
  # in the non-negative border of the SOS-cone in the 4-4 case.
  p1 := x1^2- x4^2;
  p2 := x2^2- x4^2;
  p3 := x3^2- x4^2;
  p4 := -x1^2 - x1*x2 - x1*x3 + x1*x4 - x2*x3 + x2*x4 + x3*x4;
  p5 := x1*x5 + x2*x5;
  p6 := x3*x5 + x2*x5;
```

```
p7 := x5*x5;
g := p1^2 + p2^2 + p3^2 + p4^2 + p5^2 + p6^2 + p7^2;
```

$$p1 := x1^2 - x4^2$$

$$p2 := x2^2 - x4^2$$

$$p3 := x3^2 - x4^2$$

$$p4 := -x1^2 - x1\,x2 - x1\,x3 + x1\,x4 - x2\,x3 + x2\,x4 + x3\,x4$$

$$p5 := x1\,x5 + x2\,x5$$

$$p6 := x2\,x5 + x3\,x5$$

$$p7 := x5^2$$

$$g := (x1^2 - x4^2)^2 + (x2^2 - x4^2)^2 + (x3^2 - x4^2)^2 + (-x1^2 - x1\,x2 - x1\,x3 + x1\,x4 - x2\,x3 + x2\,x4 + x3\,x4)^2 + (x1\,x5 + x2\,x5)^2 + (x2\,x5 + x3\,x5)^2 + x5^4 \quad \textbf{(12)}$$

```
> out := exactSOS(g, facial = "no") :
eig(out[3]);

# There are only 7 non-zero eigenvalues, the maximum rank in the
# spectrahedron is 7.
```

"Number of indeterminates: ", 50

"Calling numerical solver SEDUMI to find the values of the remaining indeterminates..."

"SEDUMI CALL - eig"

"An exact positive definite solution could not be found for the reduced problem."

"Computing Cholesky decomposition..."

"Cholesky decomposition failed. Matrix is not positive definite. We use LDLt decomposition."

"Matrix decomposition failed for output matrix. Please check!"

(13)

$$\begin{bmatrix} -8.83857945435899\ 10^{-11} \\ -3.94162644858582\ 10^{-16} \\ -1.90644065771863\ 10^{-16} \\ -9.91754051417336\ 10^{-17} \\ -1.82849612039939\ 10^{-17} \\ 6.15568938304500\ 10^{-18} \\ 3.72409196853760\ 10^{-16} \\ 3.31832288930197\ 10^{-9} \\ 0.888960947903563 \\ 0.999999090729938 \\ 0.999999711262153 \\ 1.00000028876112 \\ 3.00000090745845 \\ 3.89989969879448 \\ 7.21113935327868 \end{bmatrix} \quad \textbf{(13)}$$

```
> ################################################################################
  # Quadratic form that vanishes in the 7 polynomials
```

```
> ## By constructing an appropriate bilinear form we show that every polynomial in a
      decomposition is a sum of
  ## p1, p2, p3, p4, x1x5, x2x5, x3x5, x4x5, x5^2
```

```
> # We construct the list of monomials of degree 2 and 4
  d := 2 :
  polVars := [x1, x2, x3, x4, x5] :
  varSum := add(polVars[i], i = 1 ..nops(polVars)) :
  md := expand((varSum)^d) :
  cfs := coeffs(md, polVars, 'ctd') :
  print("Monomials of degree d: ", ctd);
  m2d := expand(varSum^(2 * d)) :
  cfs := coeffs(m2d, polVars, 'ct2d') :
  print("Monomials of degree 2d: ", ct2d);
```

"Monomials of degree d: ", $x1^2, x1\,x2, x1\,x3, x1\,x4, x1\,x5, x2^2, x2\,x3, x2\,x4, x2\,x5, x3^2, x3\,x4, x3\,x5, x4^2, x4\,x5, x5^2$

"Monomials of degree 2d: ", $x1^4, x1^3\,x2, x1^3\,x3, x1^3\,x4, x1^3\,x5, x1^2\,x2^2, x1^2\,x2\,x3, x1^2\,x2\,x4, x1^2\,x2\,x5, x1^2\,x3^2, x1^2\,x3\,x4, x1^2\,x3\,x5, x1^2\,x4^2, x1^2\,x4\,x5, x1^2\,x5^2, x1\,x2^3, x1\,x2^2\,x3, x1\,x2^2\,x4, x1\,x2^2\,x5, x1\,x2\,x3^2, x1\,x2\,x3\,x4, x1\,x2\,x3\,x5, x1\,x2\,x4^2, x1\,x2\,x4\,x5, x1\,x2\,x5^2, x1\,x3^3, x1\,x3^2\,x4, x1\,x3^2\,x5, x1\,x3\,x4^2, x1\,x3\,x4\,x5, x1\,x3\,x5^2, x1\,x4^3, x1\,x4^2\,x5, x1\,x4\,x5^2,$ **(14)**

$x1\,x5^3, x2^4, x2^3\,x3, x2^3\,x4, x2^3\,x5, x2^2\,x3^2, x2^2\,x3\,x4, x2^2\,x3\,x5, x2^2\,x4^2, x2^2\,x4\,x5, x2^2\,x5^2,$
$x2\,x3^3, x2\,x3^2\,x4, x2\,x3^2\,x5, x2\,x3\,x4^2, x2\,x3\,x4\,x5, x2\,x3\,x5^2, x2\,x4^3, x2\,x4^2\,x5, x2\,x4\,x5^2,$
$x2\,x5^3, x3^4, x3^3\,x4, x3^3\,x5, x3^2\,x4^2, x3^2\,x4\,x5, x3^2\,x5^2, x3\,x4^3, x3\,x4^2\,x5, x3\,x4\,x5^2, x3\,x5^3, x4^4,$
$x4^3\,x5, x4^2\,x5^2, x4\,x5^3, x5^4$

```
> # A generic square
  aInd := [a[1]] :
  for i from 2 to nops([ctd]) do
   aInd := [op(aInd), a[i]];
  end do:
  hd := add(aInd[i] * ctd[i], i = 1 ..nops([ctd]));
  hd_square := hd * hd :
  aa := getCoeffs(expand(hd_square), [ct2d]) :
```

$$hd := x1^2\,a_1 + x1\,x2\,a_2 + x1\,x3\,a_3 + x1\,x4\,a_4 + x1\,x5\,a_5 + x2^2\,a_6 + x2\,x3\,a_7 + x2\,x4\,a_8 + x2\,x5\,a_9 + x3^2\,a_{10} + x3\,x4\,a_{11} + x3\,x5\,a_{12} + x4^2\,a_{13} + x4\,x5\,a_{14} + x5^2\,a_{15} \tag{15}$$

```
> # We compute all the restrictions to phi: A4 -> R given by the seven polynomials.
  # There are 10 restrictions for each polynomial

> pList := [p1, p2, p3, p4, p5, p6, p7];
  M := [ ] :
  for j from 1 to nops(pList) do
   for i from 1 to nops([ctd]) do
    p1t := expand(pList[j] * ctd[i]);
    vec := LinearAlgebra[Transpose](getCoeffs(p1t, [ct2d]));
    if (nops(M) = 0) then
     M := ⟨vec⟩;
    else
     M := ⟨M, vec⟩;
    end if;
   end do:
  end do:
```

$$pList := \left[x1^2 - x4^2, x2^2 - x4^2, x3^2 - x4^2, -x1^2 - x1\,x2 - x1\,x3 + x1\,x4 - x2\,x3 + x2\,x4 + x3\,x4, x1\,x5 + x2\,x5, x2\,x5 + x3\,x5, x5^2\right] \tag{16}$$

```
> # We solve the system using only these restrictions
  rc := [Dimension(M)];
  nr := rc[1];
  B := Vector(nr) :
  s := LinearAlgebra[LinearSolve](M, B) :
  varss := indets(s);
  nops(varss); # 1 indeterminate left to solve
```

$$rc := [105, 70]$$
$$nr := 105$$

$$varss := \{\_t6_{11}\}$$

$$1 \tag{17}$$

```
> ew := LinearAlgebra[Transpose](s) . aa : # The linear form
  oo := polyToMatrixVars(expand(ew), aInd) :
  ooM := oo[1]; # The quadratic form
```

$$ooM := [[6\_t6_{11}, -\_t6_{11}, -\_t6_{11}, \_t6_{11}, 0, 6\_t6_{11}, -\_t6_{11}, \_t6_{11}, 0, 6\_t6_{11}, \_t6_{11}, 0, 6\_t6_{11}, 0, 0],$$
$$[-\_t6_{11}, 6\_t6_{11}, -\_t6_{11}, \_t6_{11}, 0, -\_t6_{11}, -\_t6_{11}, \_t6_{11}, 0, -\_t6_{11}, \_t6_{11}, 0, -\_t6_{11}, 0, 0],$$
$$[-\_t6_{11}, -\_t6_{11}, 6\_t6_{11}, \_t6_{11}, 0, -\_t6_{11}, -\_t6_{11}, \_t6_{11}, 0, -\_t6_{11}, \_t6_{11}, 0, -\_t6_{11}, 0, 0],$$
$$[\_t6_{11}, \_t6_{11}, \_t6_{11}, 6\_t6_{11}, 0, \_t6_{11}, \_t6_{11}, -\_t6_{11}, 0, \_t6_{11}, -\_t6_{11}, 0, \_t6_{11}, 0, 0],$$
$$[0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0],$$
$$[6\_t6_{11}, -\_t6_{11}, -\_t6_{11}, \_t6_{11}, 0, 6\_t6_{11}, -\_t6_{11}, \_t6_{11}, 0, 6\_t6_{11}, \_t6_{11}, 0, 6\_t6_{11}, 0, 0],$$
$$[-\_t6_{11}, -\_t6_{11}, -\_t6_{11}, \_t6_{11}, 0, -\_t6_{11}, 6\_t6_{11}, \_t6_{11}, 0, -\_t6_{11}, \_t6_{11}, 0, -\_t6_{11}, 0, 0],$$
$$[\_t6_{11}, \_t6_{11}, \_t6_{11}, -\_t6_{11}, 0, \_t6_{11}, \_t6_{11}, 6\_t6_{11}, 0, \_t6_{11}, -\_t6_{11}, 0, \_t6_{11}, 0, 0],$$
$$[0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0],$$
$$[6\_t6_{11}, -\_t6_{11}, -\_t6_{11}, \_t6_{11}, 0, 6\_t6_{11}, -\_t6_{11}, \_t6_{11}, 0, 6\_t6_{11}, \_t6_{11}, 0, 6\_t6_{11}, 0, 0],$$
$$[\_t6_{11}, \_t6_{11}, \_t6_{11}, -\_t6_{11}, 0, \_t6_{11}, \_t6_{11}, -\_t6_{11}, 0, \_t6_{11}, 6\_t6_{11}, 0, \_t6_{11}, 0, 0],$$
$$[0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0],$$
$$[6\_t6_{11}, -\_t6_{11}, -\_t6_{11}, \_t6_{11}, 0, 6\_t6_{11}, -\_t6_{11}, \_t6_{11}, 0, 6\_t6_{11}, \_t6_{11}, 0, 6\_t6_{11}, 0, 0],$$
$$[0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0],$$
$$[0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0]] \tag{18}$$

```
> # Only 1 indeterminate. Any positive value will give a PSD matrix with 9 elements in the kernel
  oEval := eval(ooM, {varss[1]=1}) :
  eig(oEval);
```

(19)

$$\begin{bmatrix} -3.02652253289528\ 10^{-15} \\ -1.72049927949540\ 10^{-15} \\ -7.68864566142103\ 10^{-16} \\ 0. \\ 0. \\ 5.50064098724836\ 10^{-17} \\ 9.15084076981504\ 10^{-16} \\ 2.56873183348618\ 10^{-15} \\ 9.48910401304221\ 10^{-15} \\ 7.00000000000000 \\ 7.00000000000000 \\ 7.00000000000000 \\ 7.00000000000000 \\ 7.00000000000000 \\ 25.0000000000000 \end{bmatrix} \tag{19}$$

```
> # Polynomials in the kernel
  L := NullSpace(oEval);
  ctdV := convert([ctd], Vector) :
  L[1] . ctdV;
  L[2] . ctdV;
  L[3] . ctdV;
  L[4] . ctdV;
  L[5] . ctdV;
  L[6] . ctdV;
  L[7] . ctdV;
  L[8] . ctdV;
  L[9] . ctdV;
```

$$L := \left\{ \begin{bmatrix} -1 \\ 0 \\ 0 \\ 0 \\ 0 \\ 1 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \end{bmatrix}, \begin{bmatrix} 0 \\ 0 \\ 0 \\ 0 \\ 1 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \end{bmatrix}, \begin{bmatrix} 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 1 \end{bmatrix}, \begin{bmatrix} 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 1 \\ 0 \end{bmatrix}, \begin{bmatrix} -1 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 1 \\ 0 \\ 0 \end{bmatrix}, \begin{bmatrix} 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 1 \\ 0 \\ 0 \\ 0 \end{bmatrix}, \begin{bmatrix} -1 \\ -1 \\ -1 \\ 1 \\ 0 \\ 0 \\ -1 \\ 1 \\ 0 \\ 0 \\ 1 \\ 0 \\ 0 \\ 0 \\ 0 \end{bmatrix}, \begin{bmatrix} -1 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 1 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \end{bmatrix}, \begin{bmatrix} 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 1 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \end{bmatrix} \right\}$$

$$-x1^2 + x2^2$$
$$x1\,x5$$
$$x5^2$$
$$x4\,x5$$
$$-x1^2 + x4^2$$
$$x3\,x5$$
$$-x1^2 - x1\,x2 - x1\,x3 + x1\,x4 - x2\,x3 + x2\,x4 + x3\,x4$$
$$-x1^2 + x3^2$$
$$x2\,x5 \tag{20}$$

> # The 9 polynomials in the kernel are
> # p1, p2, p3, p4, x1*x5, x2*x5, x3*x5, x4*x5, x5*x5

> *# We need to show that a sum of squares of 6 linear combinations of*
> *# these polynomials cannot give f.*

> *# The kernel is generated by p1, ...,p4, m1, ..., m5*
> $m1 := x1 * x5;$
> $m2 := x2 * x5;$
> $m3 := x3 * x5;$
> $m4 := x4 * x5;$
> $m5 := x5 * x5;$

$$m1 := x1\,x5$$
$$m2 := x2\,x5$$
$$m3 := x3\,x5$$
$$m4 := x4\,x5$$

$$m5 := x5^2 \tag{21}$$

```
> ################################## First step

# We construct the sum of squares and equate the coefficients

q1 := a11*f1 + a21*f2 + a31*f3 + a41*f4 + a51*m1 + a61*m2 + a71*m3 + a81*m4
      + a91*m5;
q2 :=      a22*f2 + a32*f3 + a42*f4 + a52*m1 + a62*m2 + a72*m3 + a82*m4 + a92
      *m5;
q3 :=            a33*f3 + a43*f4 + a53*m1 + a63*m2 + a73*m3 + a83*m4 + a93*m5;
q4 :=                  a44*f4 + a54*m1 + a64*m2 + a74*m3 + a84*m4 + a94*m5;
q5 :=                        a55*m1 + a65*m2 + a75*m3 + a85*m4 + a95*m5;
q6 :=                              a66*m2 + a76*m3 + a86*m4 + a96*m5;
q := q1^2 + q2^2 + q3^2 + q4^2 + q5^2 + q6^2;
```

$$q1 := a11\,(x1^2 - x4^2) + a21\,(x2^2 - x4^2) + a31\,(x3^2 - x4^2) + a41\,(-x1^2 - x1\,x2 - x1\,x3 + x1\,x4 - x2\,x3 + x2\,x4 + x3\,x4) + a51\,x1\,x5 + a61\,x2\,x5 + a71\,x3\,x5 + a81\,x4\,x5 + a91\,x5^2$$

$$q2 := a22\,(x2^2 - x4^2) + a32\,(x3^2 - x4^2) + a42\,(-x1^2 - x1\,x2 - x1\,x3 + x1\,x4 - x2\,x3 + x2\,x4 + x3\,x4) + a52\,x1\,x5 + a62\,x2\,x5 + a72\,x3\,x5 + a82\,x4\,x5 + a92\,x5^2$$

$$q3 := a33\,(x3^2 - x4^2) + a43\,(-x1^2 - x1\,x2 - x1\,x3 + x1\,x4 - x2\,x3 + x2\,x4 + x3\,x4) + a53\,x1\,x5 + a63\,x2\,x5 + a73\,x3\,x5 + a83\,x4\,x5 + a93\,x5^2$$

$$q4 := a44\,(-x1^2 - x1\,x2 - x1\,x3 + x1\,x4 - x2\,x3 + x2\,x4 + x3\,x4) + a54\,x1\,x5 + a64\,x2\,x5 + a74\,x3\,x5 + a84\,x4\,x5 + a94\,x5^2$$

$$q5 := a55\,x1\,x5 + a65\,x2\,x5 + a75\,x3\,x5 + a85\,x4\,x5 + a95\,x5^2$$

$$q6 := a66\,x2\,x5 + a76\,x3\,x5 + a86\,x4\,x5 + a96\,x5^2$$

$$q := \big(a11\,(x1^2 - x4^2) + a21\,(x2^2 - x4^2) + a31\,(x3^2 - x4^2) + a41\,(-x1^2 - x1\,x2 - x1\,x3 + x1\,x4 - x2\,x3 + x2\,x4 + x3\,x4) + a51\,x1\,x5 + a61\,x2\,x5 + a71\,x3\,x5 + a81\,x4\,x5 + a91\,x5^2\big)^2 + \big(a22\,(x2^2 - x4^2) + a32\,(x3^2 - x4^2) + a42\,(-x1^2 - x1\,x2 - x1\,x3 + x1\,x4 - x2\,x3 + x2\,x4 + x3\,x4) + a52\,x1\,x5 + a62\,x2\,x5 + a72\,x3\,x5 + a82\,x4\,x5 + a92\,x5^2\big)^2 + \big(a33\,(x3^2 - x4^2) + a43\,(-x1^2 - x1\,x2 - x1\,x3 + x1\,x4 - x2\,x3 + x2\,x4 + x3\,x4) + a53\,x1\,x5 + a63\,x2\,x5 + a73\,x3\,x5 + a83\,x4\,x5 + a93\,x5^2\big)^2 + \big(a44\,(-x1^2 - x1\,x2 - x1\,x3 + x1\,x4 - x2\,x3 + x2\,x4 + x3\,x4) + a54\,x1\,x5 + a64\,x2\,x5 + a74\,x3\,x5 + a84\,x4\,x5 + a94\,x5^2\big)^2 + \big(a55\,x1\,x5 + a65\,x2\,x5 + a75\,x3\,x5 + a85\,x4\,x5 + a95\,x5^2\big)^2 + \big(a66\,x2\,x5 + a76\,x3\,x5 + a86\,x4\,x5 + a96\,x5^2\big)^2 \tag{22}$$

```
> ## We copy the construction of the monomials of degree 2 and 4
## (to make this verification part independent from the previous parts)

d := 2 :
polVars := [x1, x2, x3, x4, x5] :
varSum := add(polVars[i], i = 1 ..nops(polVars)) :
```

```
md := expand( (varSum)^d ) :
cfs := coeffs(md, polVars, 'ctd') :
m2d := expand(varSum^(2 * d)) :
cfs := coeffs(m2d, polVars, 'ct2d') :

cof := getCoeffs(expand(f), [ct2d]) :
cog := getCoeffs(expand(q), [ct2d]) :
```

```
> # Equations for Singular
  # This gives a list of equations, and we compute in Singular a Groebner
  # base of the ideal, to get simpler equations, using degree reverse
  # lexicographical ordering.

  eqs := cof − cog :
  eqsList := convert(eqs, list);
```

$$eqsList := [-a11^2 + 2\,a11\,a41 - a41^2 - a42^2 - a43^2 - a44^2 + 2, 2\,a11\,a41 - 2\,a41^2 - 2\,a42^2 - 2\,a43^2 - 2\,a44^2 + 2, 2\,a11\,a41 - 2\,a41^2 - 2\,a42^2 - 2\,a43^2 - 2\,a44^2 + 2, -2\,a11\,a41 + 2\,a41^2 + 2\,a42^2 + 2\,a43^2 + 2\,a44^2 - 2, -2\,a11\,a51 + 2\,a41\,a51 + 2\,a42\,a52 + 2\,a43\,a53 + 2\,a44\,a54, -2\,a11\,a21 + 2\,a21\,a41 + 2\,a22\,a42 - a41^2 - a42^2 - a43^2 - a44^2 + 1, 2\,a11\,a41 - 4\,a41^2 - 4\,a42^2 - 4\,a43^2 - 4\,a44^2 + 4, -2\,a11\,a41 + 4\,a41^2 + 4\,a42^2 + 4\,a43^2 + 4\,a44^2 - 4, -2\,a11\,a61 + 2\,a41\,a51 + 2\,a41\,a61 + 2\,a42\,a52 + 2\,a42\,a62 + 2\,a43\,a53 + 2\,a43\,a63 + 2\,a44\,a54 + 2\,a44\,a64, -2\,a11\,a31 + 2\,a31\,a41 + 2\,a32\,a42 + 2\,a33\,a43 - a41^2 - a42^2 - a43^2 - a44^2 + 1, -2\,a11\,a41 + 4\,a41^2 + 4\,a42^2 + 4\,a43^2 + 4\,a44^2 - 4, -2\,a11\,a71 + 2\,a41\,a51 + 2\,a41\,a71 + 2\,a42\,a52 + 2\,a42\,a72 + 2\,a43\,a53 + 2\,a43\,a73 + 2\,a44\,a54 + 2\,a44\,a74, 2\,a11^2 + 2\,a11\,a21 + 2\,a11\,a31 - 2\,a11\,a41 - 2\,a21\,a41 - 2\,a22\,a42 - 2\,a31\,a41 - 2\,a32\,a42 - 2\,a33\,a43 - a41^2 - a42^2 - a43^2 - a44^2 - 1, -2\,a11\,a81 - 2\,a41\,a51 + 2\,a41\,a81 - 2\,a42\,a52 + 2\,a42\,a82 - 2\,a43\,a53 + 2\,a43\,a83 - 2\,a44\,a54 + 2\,a44\,a84, -2\,a11\,a91 + 2\,a41\,a91 + 2\,a42\,a92 + 2\,a43\,a93 + 2\,a44\,a94 - a51^2 - a52^2 - a53^2 - a54^2 - a55^2 + 1, 2\,a21\,a41 + 2\,a22\,a42, 2\,a21\,a41 + 2\,a22\,a42 - 2\,a41^2 - 2\,a42^2 - 2\,a43^2 - 2\,a44^2 + 2, -2\,a21\,a41 - 2\,a22\,a42 + 2\,a41^2 + 2\,a42^2 + 2\,a43^2 + 2\,a44^2 - 2, -2\,a21\,a51 - 2\,a22\,a52 + 2\,a41\,a61 + 2\,a42\,a62 + 2\,a43\,a63 + 2\,a44\,a64, 2\,a31\,a41 + 2\,a32\,a42 + 2\,a33\,a43 - 2\,a41^2 - 2\,a42^2 - 2\,a43^2 - 2\,a44^2 + 2, 6\,a41^2 + 6\,a42^2 + 6\,a43^2 + 6\,a44^2 - 6, 2\,a41\,a51 + 2\,a41\,a61 + 2\,a41\,a71 + 2\,a42\,a52 + 2\,a42\,a62 + 2\,a42\,a72 + 2\,a43\,a53 + 2\,a43\,a63 + 2\,a43\,a73 + 2\,a44\,a54 + 2\,a44\,a64 + 2\,a44\,a74, -2\,a11\,a41 - 2\,a21\,a41 - 2\,a22\,a42 - 2\,a31\,a41 - 2\,a32\,a42 - 2\,a33\,a43 - 2\,a41^2 - 2\,a42^2 - 2\,a43^2 - 2\,a44^2 + 2, -2\,a41\,a51 - 2\,a41\,a61 + 2\,a41\,a81 - 2\,a42\,a52 - 2\,a42\,a62 + 2\,a42\,a82 - 2\,a43\,a53 - 2\,a43\,a63 + 2\,a43\,a83 - 2\,a44\,a54 - 2\,a44\,a64 + 2\,a44\,a84, 2\,a41\,a91 + 2\,a42\,a92 + 2\,a43\,a93 + 2\,a44\,a94 - 2\,a51\,a61 - 2\,a52\,a62 - 2\,a53\,a63 - 2\,a54\,a64 - 2\,a55\,a65 + 2, 2\,a31\,a41 + 2\,a32\,a42 + 2\,a33\,a43, -2\,a31\,a41 - 2\,a32\,a42 - 2\,a33\,a43 + 2\,a41^2 + 2\,a42^2 + 2\,a43^2 + 2\,a44^2 - 2, \quad (23)$$

$-2\, a31\, a51 - 2\, a32\, a52 - 2\, a33\, a53 + 2\, a41\, a71 + 2\, a42\, a72 + 2\, a43\, a73 + 2\, a44\, a74,$
$-2\, a11\, a41 - 2\, a21\, a41 - 2\, a22\, a42 - 2\, a31\, a41 - 2\, a32\, a42 - 2\, a33\, a43 - 2\, a41^2$
$- 2\, a42^2 - 2\, a43^2 - 2\, a44^2 + 2,\ -2\, a41\, a51 - 2\, a41\, a71 + 2\, a41\, a81 - 2\, a42\, a52$
$- 2\, a42\, a72 + 2\, a42\, a82 - 2\, a43\, a53 - 2\, a43\, a73 + 2\, a43\, a83 - 2\, a44\, a54 - 2\, a44\, a74$
$+ 2\, a44\, a84, 2\, a41\, a91 + 2\, a42\, a92 + 2\, a43\, a93 + 2\, a44\, a94 - 2\, a51\, a71 - 2\, a52\, a72$
$- 2\, a53\, a73 - 2\, a54\, a74 - 2\, a55\, a75, 2\, a11\, a41 + 2\, a21\, a41 + 2\, a22\, a42 + 2\, a31\, a41$
$+ 2\, a32\, a42 + 2\, a33\, a43, 2\, a11\, a51 + 2\, a21\, a51 + 2\, a22\, a52 + 2\, a31\, a51 + 2\, a32\, a52$
$+ 2\, a33\, a53 - 2\, a41\, a81 - 2\, a42\, a82 - 2\, a43\, a83 - 2\, a44\, a84,\ -2\, a41\, a91 - 2\, a42\, a92$
$- 2\, a43\, a93 - 2\, a44\, a94 - 2\, a51\, a81 - 2\, a52\, a82 - 2\, a53\, a83 - 2\, a54\, a84$
$- 2\, a55\, a85,\ -2\, a51\, a91 - 2\, a52\, a92 - 2\, a53\, a93 - 2\, a54\, a94 - 2\, a55\, a95,\ -a21^2$
$- a22^2 + 1, 2\, a21\, a41 + 2\, a22\, a42,\ -2\, a21\, a41 - 2\, a22\, a42,\ -2\, a21\, a61 - 2\, a22\, a62,$
$-2\, a21\, a31 - 2\, a22\, a32 - a41^2 - a42^2 - a43^2 - a44^2 + 1,\ -2\, a21\, a41 - 2\, a22\, a42$
$+ 2\, a41^2 + 2\, a42^2 + 2\, a43^2 + 2\, a44^2 - 2,\ -2\, a21\, a71 - 2\, a22\, a72 + 2\, a41\, a61$
$+ 2\, a42\, a62 + 2\, a43\, a63 + 2\, a44\, a64, 2\, a11\, a21 + 2\, a21^2 + 2\, a21\, a31 + 2\, a22^2$
$+ 2\, a22\, a32 - a41^2 - a42^2 - a43^2 - a44^2 - 1,\ -2\, a21\, a81 - 2\, a22\, a82 - 2\, a41\, a61$
$- 2\, a42\, a62 - 2\, a43\, a63 - 2\, a44\, a64,\ -2\, a21\, a91 - 2\, a22\, a92 - a61^2 - a62^2 - a63^2$
$- a64^2 - a65^2 - a66^2 + 2, 2\, a31\, a41 + 2\, a32\, a42 + 2\, a33\, a43,\ -2\, a31\, a41 - 2\, a32\, a42$
$- 2\, a33\, a43 + 2\, a41^2 + 2\, a42^2 + 2\, a43^2 + 2\, a44^2 - 2,\ -2\, a31\, a61 - 2\, a32\, a62$
$- 2\, a33\, a63 + 2\, a41\, a71 + 2\, a42\, a72 + 2\, a43\, a73 + 2\, a44\, a74,\ -2\, a11\, a41 - 2\, a21\, a41$
$- 2\, a22\, a42 - 2\, a31\, a41 - 2\, a32\, a42 - 2\, a33\, a43 - 2\, a41^2 - 2\, a42^2 - 2\, a43^2 - 2\, a44^2$
$+ 2,\ -2\, a41\, a61 - 2\, a41\, a71 + 2\, a41\, a81 - 2\, a42\, a62 - 2\, a42\, a72 + 2\, a42\, a82$
$- 2\, a43\, a63 - 2\, a43\, a73 + 2\, a43\, a83 - 2\, a44\, a64 - 2\, a44\, a74 + 2\, a44\, a84, 2\, a41\, a91$
$+ 2\, a42\, a92 + 2\, a43\, a93 + 2\, a44\, a94 - 2\, a61\, a71 - 2\, a62\, a72 - 2\, a63\, a73 - 2\, a64\, a74$
$- 2\, a65\, a75 - 2\, a66\, a76 + 2, 2\, a11\, a41 + 2\, a21\, a41 + 2\, a22\, a42 + 2\, a31\, a41$
$+ 2\, a32\, a42 + 2\, a33\, a43, 2\, a11\, a61 + 2\, a21\, a61 + 2\, a22\, a62 + 2\, a31\, a61 + 2\, a32\, a62$
$+ 2\, a33\, a63 - 2\, a41\, a81 - 2\, a42\, a82 - 2\, a43\, a83 - 2\, a44\, a84,\ -2\, a41\, a91 - 2\, a42\, a92$
$- 2\, a43\, a93 - 2\, a44\, a94 - 2\, a61\, a81 - 2\, a62\, a82 - 2\, a63\, a83 - 2\, a64\, a84 - 2\, a65\, a85$
$- 2\, a66\, a86,\ -2\, a61\, a91 - 2\, a62\, a92 - 2\, a63\, a93 - 2\, a64\, a94 - 2\, a65\, a95$
$- 2\, a66\, a96,\ -a31^2 - a32^2 - a33^2 + 1,\ -2\, a31\, a41 - 2\, a32\, a42 - 2\, a33\, a43,\ -2\, a31\, a71$
$- 2\, a32\, a72 - 2\, a33\, a73, 2\, a11\, a31 + 2\, a21\, a31 + 2\, a22\, a32 + 2\, a31^2 + 2\, a32^2 + 2\, a33^2$
$- a41^2 - a42^2 - a43^2 - a44^2 - 1,\ -2\, a31\, a81 - 2\, a32\, a82 - 2\, a33\, a83 - 2\, a41\, a71$
$- 2\, a42\, a72 - 2\, a43\, a73 - 2\, a44\, a74,\ -2\, a31\, a91 - 2\, a32\, a92 - 2\, a33\, a93 - a71^2$
$- a72^2 - a73^2 - a74^2 - a75^2 - a76^2 + 1, 2\, a11\, a41 + 2\, a21\, a41 + 2\, a22\, a42$
$+ 2\, a31\, a41 + 2\, a32\, a42 + 2\, a33\, a43, 2\, a11\, a71 + 2\, a21\, a71 + 2\, a22\, a72 + 2\, a31\, a71$
$+ 2\, a32\, a72 + 2\, a33\, a73 - 2\, a41\, a81 - 2\, a42\, a82 - 2\, a43\, a83 - 2\, a44\, a84,\ -2\, a41\, a91$
$- 2\, a42\, a92 - 2\, a43\, a93 - 2\, a44\, a94 - 2\, a71\, a81 - 2\, a72\, a82 - 2\, a73\, a83 - 2\, a74\, a84$

$-2\,a75\,a85-2\,a76\,a86$, $-2\,a71\,a91-2\,a72\,a92-2\,a73\,a93-2\,a74\,a94-2\,a75\,a95-2\,a76\,a96$, $-a11^2-2\,a11\,a21-2\,a11\,a31-a21^2-2\,a21\,a31-a22^2-2\,a22\,a32-a31^2-a32^2-a33^2+3$, $2\,a11\,a81+2\,a21\,a81+2\,a22\,a82+2\,a31\,a81+2\,a32\,a82+2\,a33\,a83$, $2\,a11\,a91+2\,a21\,a91+2\,a22\,a92+2\,a31\,a91+2\,a32\,a92+2\,a33\,a93-a81^2-a82^2-a83^2-a84^2-a85^2-a86^2$, $-2\,a81\,a91-2\,a82\,a92-2\,a83\,a93-2\,a84\,a94-2\,a85\,a95-2\,a86\,a96$, $-a91^2-a92^2-a93^2-a94^2-a95^2-a96^2+1\,]$

> # See the Singular file for this step.

> ################################## Second step

```
# We get this polynomials in the Greobner basis (among others)
#J2[1]=a96
#J2[2]=a95
#J2[3]=a84
#J2[4]=a83
#J2[5]=a82
#J2[6]=a81
#J2[7]=a75+a85
#J2[8]=a74
#J2[9]=a64
#J2[10]=a54
#J2[11]=a73
#J2[12]=a63
#J2[13]=a53
#J2[14]=a43
#J2[15]=a72
#J2[16]=a62
#J2[17]=a52
#J2[18]=a42
#J2[19]=a32
#J2[20]=a71
#J2[21]=a61
#J2[22]=a51
#J2[23]=a41
#J2[24]=a31
#J2[25]=a21
#J2[38]=a44^2-1
#J2[39]=a33^2-1
#J2[40]=a22^2-1
#J2[41]=a11^2-1
```

> # From Singular GB we deduce that we can assume a11 = a22 = a33 = a44 = 1.

(The coefficients must be real, and p^2 = (-p)^2 so we can set the initial coefficient to be positive.)

> # Moreover, we take

a96 = a95 = a84 = a83 = a82 = a81 = a74 = a64 = a54 = a73 = a63 = a53 = a43 = a72

$= a62 = a52 = a42 = a32 = a71 = a61 = a51 = a41 = a31 = a21 = 0$

```
# a75 = a85
```

```
> # The new system is of the shape
q1 := f1                                   + a91 * m5;
q2 :=       f2                             + a92 * m5;
q3 :=             f3                       + a93 * m5;
q4 :=                   f4                 + a94 * m5;
q5 :=        a55 * m1 + a65 * m2 + a75 * m3 - a75 * m4    ;
q6 :=                   a66 * m2 + a76 * m3 + a86 * m4    ;
g := q1^2 + q2^2 + q3^2 + q4^2 + q5^2 + q6^2;
```

$$q1 := a91\,x5^2 + x1^2 - x4^2$$
$$q2 := a92\,x5^2 + x2^2 - x4^2$$
$$q3 := a93\,x5^2 + x3^2 - x4^2$$
$$q4 := a94\,x5^2 - x1^2 - x1\,x2 - x1\,x3 + x1\,x4 - x2\,x3 + x2\,x4 + x3\,x4$$
$$q5 := a55\,x1\,x5 + a65\,x2\,x5 + a75\,x3\,x5 - a75\,x4\,x5$$
$$q6 := a66\,x2\,x5 + a76\,x3\,x5 + a86\,x4\,x5$$
$$g := \left(a91\,x5^2 + x1^2 - x4^2\right)^2 + \left(a92\,x5^2 + x2^2 - x4^2\right)^2 + \left(a93\,x5^2 + x3^2 - x4^2\right)^2 + \left(a94\,x5^2 - x1^2 - x1\,x2 - x1\,x3 + x1\,x4 - x2\,x3 + x2\,x4 + x3\,x4\right)^2 + \left(a55\,x1\,x5 + a65\,x2\,x5 + a75\,x3\,x5 - a75\,x4\,x5\right)^2 + \left(a66\,x2\,x5 + a76\,x3\,x5 + a86\,x4\,x5\right)^2 \tag{24}$$

```
> cof := getCoeffs(expand(f), [ct2d]) :
cog := getCoeffs(expand(g), [ct2d]) :
```

```
> # List of equations to solve (in Singular)
eqs := cof - cog :
eqsList := convert(eqs, list);
```

$$eqsList := [0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, -a55^2 - 2\,a91 + 2\,a94 + 1, 0, 0, 0, 0, 0, 0, 0, 0, 0, -2\,a55\,a65 + 2\,a94 + 2, 0, 0, 0, 0, 0, -2\,a55\,a75 + 2\,a94, 0, 0, 2\,a55\,a75 - 2\,a94, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, -a65^2 - a66^2 - 2\,a92 + 2, 0, 0, 0, 0, 0, -2\,a65\,a75 - 2\,a66\,a76 + 2\,a94 + 2, 0, 0, 2\,a65\,a75 - 2\,a66\,a86 - 2\,a94, 0, 0, 0, 0, 0, 0, -a75^2 - a76^2 - 2\,a93 + 1, 0, 0, 2\,a75^2 - 2\,a76\,a86 - 2\,a94, 0, 0, 0, -a75^2 - a86^2 + 2\,a91 + 2\,a92 + 2\,a93, 0, -a91^2 - a92^2 - a93^2 - a94^2 + 1] \tag{25}$$

```
> # We use now factorizing GB so that we can split the ideal and get simpler equations.
# We get three components. The following polynomials are in the components:
```

```
> # Component 1

    # -88305*a94^8-195112*a94^7-23548*a94^6+262392*a94^5+195112*a94^4-87464*
    a94^3-171564*a94^2-80736*a94-13456

evalf(allvalues(RootOf(-88305*a94^8-195112*a94^7-23548*a94^6+262392*a94^5
    +195112*a94^4-87464*a94^3-171564*a94^2-80736*a94-13456)))
```

#.9770774009+.3039278136*I, -.4880660791+.2369050395*I, -.7886159441+.3702161294*I, -.8051572825+0.5652380060e-1*I, -.8051572825 -0.5652380060e-1*I, -.7886159441-.3702161294*I, -.4880660791-.2369050395*I, .9770774009-.3039278136*I

$$0.9770774009 + 0.3039278136\, I,\ -0.4880660791 + 0.2369050395\, I,\ -0.7886159441 + 0.3702161294\, I,\ -0.8051572825 + 0.05652380060\, I,\ -0.8051572825 - 0.05652380060\, I,\ -0.7886159441 - 0.3702161294\, I,\ -0.4880660791 - 0.2369050395\, I,\ 0.9770774009 - 0.3039278136\, I \tag{26}$$

```
> # Component 2
  # 16*a93^6+8*a93^5-12*a93^4+4*a93^3+4*a93^2-4*a93+1

  evalf(allvalues(RootOf(16*a93^6+8*a93^5-12*a93^4+4*a93^3+4*a93^2-4*a93
    +1)))

    # .4147417705+.1266183578*I, .2932789419+.5202065681*I,
    -.9580207124+.1190792894*I, -.9580207124-.1190792894*I, .2932789419-.5202065681*I,
    .4147417705-.1266183578*I
```

$$0.4147417705 + 0.1266183578\, I,\ 0.2932789419 + 0.5202065681\, I,\ -0.9580207124 + 0.1190792894\, I,\ -0.9580207124 - 0.1190792894\, I,\ 0.2932789419 - 0.5202065681\, I,\ 0.4147417705 - 0.1266183578\, I \tag{27}$$

```
> # Component 3
  # 392*a94^6-224*a94^5+32*a94^3+80*a94^2+16

  evalf(allvalues(RootOf(392*a94^6-224*a94^5+32*a94^3+80*a94^2+16)))

    # .6950538153+.4666975070*I, 0.4575932903e-1+.4009387055*I,
    -.4550988586+.3879239341*I, -.4550988586-.3879239341*I, 0.4575932903e-1
    -.4009387055*I, .6950538153-.4666975070*I
```

$$0.6950538153 + 0.4666975070\, I,\ 0.04575932903 + 0.4009387055\, I,\ -0.4550988586 + 0.3879239341\, I,\ -0.4550988586 - 0.3879239341\, I,\ 0.04575932903 - 0.4009387055\, I,\ 0.6950538153 - 0.4666975070\, I \tag{28}$$

```
> # Non of these polynomials have real roots, so there is no real solution for the problem.
  # f cannot be decomposed as sum of 6 squares.

>
```