```
> # Construction of strictly positive polynomial of degree 8 in 3 variables in the border of the sum
      of squares cone, that is the sum of 5 squares.
  # See the details in Section 4.3.1 of "Strictly positive polynomials in the border of the SOS cone",
      by S. Laplagne and M. Valdettaro.
> ##########################################################################
  # Load "Rational SOS" procedures
  ##########################################################################
  read("rationalSOS.mpl") :
  with(rationalSOS) :
  with(LinearAlgebra) :

  # Display tables of any size
  interface(rtablesize = infinity);
```

"Opening connection with Matlab"

10 **(1)**

```
> ##########################################################################
   # Construction of a polyonmial of degree 8 in the border

   # We define two polynomials with 16 common roots
  p1 := (x1) * (x1-x0) * (x1-2 * x0) * (x1 + x0);
  p2 := (x2) * (x2-x0) * (x2-2 * x0) * (x2 + x0);

  # The list of common roots.
  sols := solve({p1, p2, x0-1});
  nops([sols]); # 16 solutions
```

$$p1 := x1\ (x1 - x0)\ (x1 - 2\ x0)\ (x1 + x0)$$

$$p2 := x2\ (x2 - x0)\ (x2 - 2\ x0)\ (x2 + x0)$$

$sols := \{x0 = 1, x1 = 0, x2 = 0\}, \{x0 = 1, x1 = 0, x2 = 1\}, \{x0 = 1, x1 = 0, x2 = 2\}, \{x0 = 1, x1 = 0, x2 = -1\}, \{x0 = 1, x1 = 1, x2 = 0\}, \{x0 = 1, x1 = 2, x2 = 0\}, \{x0 = 1, x1 = -1, x2 = 0\}, \{x0 = 1, x1 = 1, x2 = 1\}, \{x0 = 1, x1 = 1, x2 = 2\}, \{x0 = 1, x1 = 1, x2 = -1\}, \{x0 = 1, x1 = 2, x2 = 1\}, \{x0 = 1, x1 = -1, x2 = 1\}, \{x0 = 1, x1 = -1, x2 = 2\}, \{x0 = 1, x1 = -1, x2 = -1\}, \{x0 = 1, x1 = 2, x2 = 2\}, \{x0 = 1, x1 = 2, x2 = -1\}$

16 **(2)**

```
> ##########################################################################
  # We construct the quadratic form
  ##########################################################################
> # We define the set of monomials of degree d (stored in ctd) and
  # the set of monomials of degree 2d (stored in ct2d), for d = 4.
  d := 4 :
  polVars := [x0, x1, x2] :
  varSum := add(polVars[i], i = 1 ..nops(polVars)) :
  md := expand((varSum)^d) :
  cfs := coeffs(md, polVars, 'ctd') :
  print("Monomials of degree d: ", ctd);
  m2d := expand(varSum^(2 * d)) :
```

```
cfs := coeffs(m2d, polVars, 'ct2d') :
print("Monomials of degree 2d: ", ct2d);
```

"Monomials of degree d: ", $x0^4, x0^3\,x1, x0^3\,x2, x0^2\,x1^2, x0^2\,x1\,x2, x0^2\,x2^2, x0\,x1^3, x0\,x1^2\,x2, x0\,x1\,x2^2, x0\,x2^3, x1^4, x1^3\,x2, x1^2\,x2^2, x1\,x2^3, x2^4$

"Monomials of degree 2d: ", $x0^8, x0^7\,x1, x0^7\,x2, x0^6\,x1^2, x0^6\,x1\,x2, x0^6\,x2^2, x0^5\,x1^3, x0^5\,x1^2\,x2, x0^5\,x1\,x2^2, x0^5\,x2^3, x0^4\,x1^4, x0^4\,x1^3\,x2, x0^4\,x1^2\,x2^2, x0^4\,x1\,x2^3, x0^4\,x2^4, x0^3\,x1^5, x0^3\,x1^4\,x2, x0^3\,x1^3\,x2^2, x0^3\,x1^2\,x2^3, x0^3\,x1\,x2^4, x0^3\,x2^5, x0^2\,x1^6, x0^2\,x1^5\,x2, x0^2\,x1^4\,x2^2, x0^2\,x1^3\,x2^3, x0^2\,x1^2\,x2^4, x0^2\,x1\,x2^5, x0^2\,x2^6, x0\,x1^7, x0\,x1^6\,x2, x0\,x1^5\,x2^2, x0\,x1^4\,x2^3, x0\,x1^3\,x2^4, x0\,x1^2\,x2^5, x0\,x1\,x2^6, x0\,x2^7, x1^8, x1^7\,x2, x1^6\,x2^2, x1^5\,x2^3, x1^4\,x2^4, x1^3\,x2^5, x1^2\,x2^6, x1\,x2^7, x2^8$ **(3)**

```
> # We define a generic polynomial of degree d with coefficientes h_i.
hCoeff := [h[1]] :
for i from 2 to nops([ctd]) do
  hCoeff := [op(hCoeff), h[i]] :
end do:
hd := add(hCoeff[i] * ctd[i], i = 1 ..nops(hCoeff)) :
print("Generic polynomial h of degree d: ", hd);
```

"Generic polynomial h of degree d: ", $x0^4\,h_1 + x0^3\,x1\,h_2 + x0^3\,x2\,h_3 + x0^2\,x1^2\,h_4 + x0^2\,x1\,x2\,h_5 + x0^2\,x2^2\,h_6 + x0\,x1^3\,h_7 + x0\,x1^2\,x2\,h_8 + x0\,x1\,x2^2\,h_9 + x0\,x2^3\,h_{10} + x1^4\,h_{11} + x1^3\,x2\,h_{12} + x1^2\,x2^2\,h_{13} + x1\,x2^3\,h_{14} + x2^4\,h_{15}$ **(4)**

```
> # We can compute the space of linear relations among the polynomials,
# and look for a relation with maximal number of null coefficients.
# We evaluate the generic polynomial in the 16 roots.
nRoots := 16 :
nCoeff := 15 : # Monomials in H_(3,4)
 alphaSeq := seq(eval(hd, sols[i]), i = 1 ..nRoots) :
```

```
> MEval := Matrix(nRoots, nCoeff) :
for i from 1 to nRoots do:
  aaC := getCoeffs(expand(alphaSeq[i]), hCoeff);
  MEval[i, 1 ..nCoeff] := aaC :
end:
u := NullSpace(Transpose(MEval));
```

$$u := \left\{ \begin{bmatrix} 3 \\ -6 \\ 3 \\ 0 \\ -3 \\ 1 \\ -1 \\ 6 \\ -3 \\ 0 \\ -2 \\ 2 \\ -1 \\ 0 \\ 1 \\ 0 \end{bmatrix}, \begin{bmatrix} 6 \\ -6 \\ 2 \\ -2 \\ -3 \\ 0 \\ -3 \\ 3 \\ -1 \\ 1 \\ 0 \\ 3 \\ -1 \\ 1 \\ 0 \\ 0 \end{bmatrix}, \begin{bmatrix} 0 \\ -3 \\ 2 \\ 1 \\ 3 \\ -2 \\ -1 \\ 0 \\ -1 \\ -2 \\ 1 \\ 2 \\ -1 \\ 0 \\ 0 \\ 1 \end{bmatrix} \right\} \tag{5}$$

```
> # We obtain a space of dimension 3, where all the generators have
  # 4 null coefficients, we can take any of these generators as a minimal
  # dependence relation.

  # Instead, following Section 4.4, we consider the set of 12 roots
  # (1:a:b) with a in {-1, 0, 1} and b in {0, 1, 2, 3}.

  nRoots := 12;
  solsSubset := [sols[1], sols[2], sols[3], sols[4], sols[5], sols[7], sols[8], sols[9], sols[10],
      sols[12], sols[13], sols[14]];
  alphaSeq := seq(eval(hd, solsSubset[i]), i = 1 ..nRoots) :
```

$$nRoots := 12$$

$$solsSubset := [\{x0=1, x1=0, x2=0\}, \{x0=1, x1=0, x2=1\}, \{x0=1, x1=0, x2=2\}, \{x0=1, x1=0, x2=-1\}, \{x0=1, x1=1, x2=0\}, \{x0=1, x1=-1, x2=0\}, \{x0=1, x1=1, x2=1\}, \{x0=1, x1=1, x2=2\}, \{x0=1, x1=1, x2=-1\}, \{x0=1, x1=-1, x2=1\}, \{x0=1, x1=-1, x2=2\}, \{x0=1, x1=-1, x2=-1\}] \tag{6}$$

```
> # There is an unique relationship with all nonzero coefficients
  MEval := Matrix(nRoots, nCoeff) :
  for i from 1 to nRoots do:
   aaC := getCoeffs(expand(alphaSeq[i]), hCoeff);
   MEval[i, 1 ..nCoeff] := aaC :
  end:
  u := NullSpace(Transpose(MEval));
  uVec := u[1] :
```

$$u := \left\{ \begin{bmatrix} 6 \\ -6 \\ 2 \\ -2 \\ -3 \\ -3 \\ 3 \\ -1 \\ 1 \\ 3 \\ -1 \\ 1 \end{bmatrix} \right\} \tag{7}$$

```
> ##############################################################################
  # We construct quadratic form Q and the associated matrix.
  ##############################################################################
```

```
> # The coefficients of the linear form l.
  # We will leave the last coefficient a_12 as indeterminate and
  # compute it using Maple to verify the theoretical formulas.
  a := [seq(uVec[i]^2, i = 1 ..nRoots)] :
  a[nRoots] := cc :
  print(a);
```

$$[36, 36, 4, 4, 9, 9, 9, 1, 1, 9, 1, cc] \tag{8}$$

```
> # We define the quadratic form Q(h), as a linear combination of
  # evaluations of h^2 in the 12 points defined above, with
  # coefficients a_i
  hd_square := expand(hd^2) :
  QForm := add(a[i]* eval(hd_square, solsSubset[i]), i = 1 ..nRoots) :
```

```
> # We construct the matrix associated to Q.
  # We define it as a 15x15 matrix with indeterminate entries, and
  # compute the entries so that c' Qmatrix c = Qform, where
  # c are the monomials of degree d.
  mSize := nCoeff:
  MM := Matrix(mSize) :
  for i to mSize do
    for j from i to mSize do
      MM[i, j] := c[i, j];
      MM[j, i] := c[i, j];
    end do:
  end do:
```

```
> # Vector q of monomials of degree d, with generic coefficients
hCoeffVector := Vector(hCoeff) :

> # We compute h' * MM * h
hCoeffVector_T := Transpose(hCoeffVector) :
hTMh := expand(hCoeffVector_T . MM . hCoeffVector) :

> # Finally we equate the coefficients of h'*MM*h and those of QForm
# and compute the coefficients of MM.
eqs := {coeffs(collect(hTMh-QForm, hCoeff,'distributed'), hCoeff) } :
sol := solve(eqs) :

> # We replace the coefficients by the values obtained
MMC := eval(MM, sol);
```

$$MMC := [[119+cc, 1-cc, 61-cc, 39+cc, -1+cc, 83+cc, 1-cc, 21-cc, 1-cc, 97-cc, 39+cc, -1+cc, 27+cc, -1+cc, 155+cc],$$
$$[1-cc, 39+cc, -1+cc, 1-cc, 21-cc, 1-cc, 39+cc, -1+cc, 27+cc, -1+cc, 1-cc, 21-cc, 1-cc, 33-cc, 1-cc],$$
$$[61-cc, -1+cc, 83+cc, 21-cc, 1-cc, 97-cc, -1+cc, 27+cc, -1+cc, 155+cc, 21-cc, 1-cc, 33-cc, 1-cc, 241-cc],$$
$$[39+cc, 1-cc, 21-cc, 39+cc, -1+cc, 27+cc, 1-cc, 21-cc, 1-cc, 33-cc, 39+cc, -1+cc, 27+cc, -1+cc, 51+cc],$$
$$[-1+cc, 21-cc, 1-cc, -1+cc, 27+cc, -1+cc, 21-cc, 1-cc, 33-cc, 1-cc, -1+cc, 27+cc, -1+cc, 51+cc, -1+cc],$$
$$[83+cc, 1-cc, 97-cc, 27+cc, -1+cc, 155+cc, 1-cc, 33-cc, 1-cc, 241-cc, 27+cc, -1+cc, 51+cc, -1+cc, 443+cc],$$
$$[1-cc, 39+cc, -1+cc, 1-cc, 21-cc, 1-cc, 39+cc, -1+cc, 27+cc, -1+cc, 1-cc, 21-cc, 1-cc, 33-cc, 1-cc],$$
$$[21-cc, -1+cc, 27+cc, 21-cc, 1-cc, 33-cc, -1+cc, 27+cc, -1+cc, 51+cc, 21-cc, 1-cc, 33-cc, 1-cc, 81-cc],$$
$$[1-cc, 27+cc, -1+cc, 1-cc, 33-cc, 1-cc, 27+cc, -1+cc, 51+cc, -1+cc, 1-cc, 33-cc, 1-cc, 81-cc, 1-cc],$$
$$[97-cc, -1+cc, 155+cc, 33-cc, 1-cc, 241-cc, -1+cc, 51+cc, -1+cc, 443+cc, 33-cc, 1-cc, 81-cc, 1-cc, 817-cc],$$
$$[39+cc, 1-cc, 21-cc, 39+cc, -1+cc, 27+cc, 1-cc, 21-cc, 1-cc, 33-cc, 39+cc, -1+cc, 27+cc, -1+cc, 51+cc],$$
$$[-1+cc, 21-cc, 1-cc, -1+cc, 27+cc, -1+cc, 21-cc, 1-cc, 33-cc, 1-cc, -1+cc, 27+cc, -1+cc, 51+cc, -1+cc],$$
$$[27+cc, 1-cc, 33-cc, 27+cc, -1+cc, 51+cc, 1-cc, 33-cc, 1-cc, 81-cc, 27+cc, -1+cc, 51+cc, -1+cc, 147+cc],$$
$$[-1+cc, 33-cc, 1-cc, -1+cc, 51+cc, -1+cc, 33-cc, 1-cc, 81-cc, 1-cc, -1+cc, 51+cc, -1+cc, 147+cc, -1+cc],$$
$$[155+cc, 1-cc, 241-cc, 51+cc, -1+cc, 443+cc, 1-cc, 81-cc, 1-cc, 817-cc, \tag{9}$$

$51 + cc, -1 + cc, 147 + cc, -1 + cc, 1595 + cc]]$

```
> ################################################################################
  # We compute the value of cc so that the kernel has dimension 5
  ################################################################################
```

```
> # The first four eigenvalues are 0.
  ev := Eigenvalues(MMC);
```

$$ev := [[0], [0], [0], [0], [RootOf(\_Z^{11} + (-2881 - 15\,cc)\,\_Z^{10} + (41632\,cc + 1599120)\,\_Z^{9} + (-21929328\,cc - 361997904)\,\_Z^{8} + (4618869120\,cc + 40330475136)\,\_Z^{7} + (-467829955584\,cc - 2424448806912)\,\_Z^{6} + (25009879375872\,cc + 81194847068160)\,\_Z^{5} + (-731808422461440\,cc - 1504671497256960)\,\_Z^{4} + (11759345813422080\,cc + 14978383141404672)\,\_Z^{3} + (-100380802384134144\,cc - 74110626925903872)\,\_Z^{2} + (414304439529111552\,cc + 148051847943290880)\,\_Z - 635539639951687680\,cc - 57776330904698880)]] \qquad (10)$$

```
> # There are 4 null eigenvalues and the remainig 11 are roots of a
  # polynomial of degree 11.
  # We choose cc so that this polynomial has a root equal to 0.
  e5 := op(ev[5]) :
  e50 := eval(e5, {_Z=0}) :
  fac := factors(e50) :
  rr := solve(fac[2][1][1]);
```

$$rr := -\frac{1}{11} \qquad (11)$$

```
> # We recover the value for a_12 predicted by the theoretical formula.
```

```
> ################################################################################
  # The five polynomials in the kernel
  ################################################################################

  MMC2 := eval(MMC, {cc=rr}) :
  nspace := NullSpace(MMC2);

  # Polynomials
  w1 := LinearAlgebra[DotProduct](nspace[1], Vector([ctd]));
  w2 := LinearAlgebra[DotProduct](nspace[2], Vector([ctd]));
  w3 := LinearAlgebra[DotProduct](nspace[3], Vector([ctd]));
  w4 := LinearAlgebra[DotProduct](nspace[4], Vector([ctd]));
  w5 := LinearAlgebra[DotProduct](nspace[5], Vector([ctd]));
```

$$nspace := \left\{ \begin{bmatrix} 0\\0\\2\\0\\0\\-1\\0\\0\\0\\-2\\0\\0\\0\\0\\1 \end{bmatrix}, \begin{bmatrix} -\frac{1}{6}\\0\\\frac{1}{6}\\\frac{1}{2}\\2\\\frac{1}{2}\\0\\-\frac{5}{2}\\-3\\-\frac{1}{3}\\0\\0\\\frac{3}{2}\\1\\0 \end{bmatrix}, \begin{bmatrix} 0\\0\\0\\0\\-1\\0\\0\\0\\0\\0\\0\\1\\0\\0\\0 \end{bmatrix}, \begin{bmatrix} 0\\0\\0\\-1\\0\\0\\0\\0\\0\\0\\1\\0\\0\\0\\0 \end{bmatrix}, \begin{bmatrix} 0\\-1\\0\\0\\0\\0\\1\\0\\0\\0\\0\\0\\0\\0\\0 \end{bmatrix} \right\}$$

$$w1 := 2\, x0^3\, x2 - x0^2\, x2^2 - 2\, x0\, x2^3 + x2^4$$

$$w2 := -\frac{1}{6}\, x0^4 + \frac{1}{6}\, x0^3\, x2 + \frac{1}{2}\, x0^2\, x1^2 + 2\, x0^2\, x1\, x2 + \frac{1}{2}\, x0^2\, x2^2 - \frac{5}{2}\, x0\, x1^2\, x2 - 3\, x0\, x1\, x2^2 - \frac{1}{3}\, x0\, x2^3 + \frac{3}{2}\, x1^2\, x2^2 + x1\, x2^3$$

$$w3 := -x0^2\, x1\, x2 + x1^3\, x2$$

$$w4 := -x0^2\, x1^2 + x1^4$$

$$w5 := -x0^3\, x1 + x0\, x1^3 \tag{12}$$

> *# We construct the sum of squares*
> $pp := w1\hat{}2 + w2\hat{}2 + w3\hat{}2 + w4\hat{}2 + w5\hat{}2;$

$$pp := \left(2\, x0^3\, x2 - x0^2\, x2^2 - 2\, x0\, x2^3 + x2^4\right)^2 + \left(-\frac{1}{6}\, x0^4 + \frac{1}{6}\, x0^3\, x2 + \frac{1}{2}\, x0^2\, x1^2 + 2\, x0^2\, x1\, x2 + \frac{1}{2}\, x0^2\, x2^2 - \frac{5}{2}\, x0\, x1^2\, x2 - 3\, x0\, x1\, x2^2 - \frac{1}{3}\, x0\, x2^3 + \frac{3}{2}\, x1^2\, x2^2 + x1\, x2^3\right)^2 + \left(-x0^2\, x1\, x2 + x1^3\, x2\right)^2 + \left(-x0^2\, x1^2 + x1^4\right)^2 + \left(-x0^3\, x1 + x0\, x1^3\right)^2 \tag{13}$$

> \# *And verify using SEDUMI that this polynomial is in the border of*
> \# *the SOS cone.*
> $out := exactSOS(pp, facial = \text{"no"}, objFunction = \text{"eig"}):$
> $eig(out[3]);$

"Number of indeterminates: ", 75

"Calling numerical solver SEDUMI to find the values of the remaining indeterminates..."

"SEDUMI CALL - eig"

"An exact positive definite solution could not be found for the reduced problem."

"Computing Cholesky decomposition..."

"Cholesky decomposition failed. Matrix is not positive definite. We use LDLt decomposition."

$$\begin{bmatrix} -3.25524975648396\ 10^{-9} \\ -1.03178163994362\ 10^{-9} \\ -9.57552137707317\ 10^{-10} \\ -6.78271457039050\ 10^{-10} \\ -2.18035694148725\ 10^{-10} \\ -2.00218600416333\ 10^{-12} \\ 3.89129412012727\ 10^{-11} \\ 4.27905722372461\ 10^{-10} \\ 6.74736303418107\ 10^{-10} \\ 1.32976450085246\ 10^{-9} \\ 1.80083493980343 \\ 1.99999444521201 \\ 2.00000496470096 \\ 9.98177451574765 \\ 23.3840859663875 \end{bmatrix} \tag{14}$$

> \# *We obtain a sum of 5 squares in the border (the matrix has rank 5).*

> \# *Moreover, the decomposition seems to be unique,*
> \# *minimizing another function we obtain the same matrix.*
> $out := exactSOS(pp, facial = \text{"no"}, objFunction = \text{"random"}):$
> $eig(out[3]);$

"Number of indeterminates: ", 75

"Calling numerical solver SEDUMI to find the values of the remaining indeterminates..."

"SEDUMI CALL - random"

"An exact positive definite solution could not be found for the reduced problem."

"Computing Cholesky decomposition..."

"Cholesky decomposition failed. Matrix is not positive definite. We use LDLt decomposition."

$$\begin{bmatrix} -5.82754725094704 \; 10^{-9} \\ -2.06384117059082 \; 10^{-9} \\ 2.46049591126366 \; 10^{-10} \\ 6.45037697978218 \; 10^{-9} \\ 1.16852775685647 \; 10^{-8} \\ 1.85444686909836 \; 10^{-8} \\ 3.76891592868326 \; 10^{-8} \\ 5.11524866793639 \; 10^{-8} \\ 1.15643056808834 \; 10^{-7} \\ 2.97798195195915 \; 10^{-7} \\ 1.79839297816478 \\ 1.99858152478882 \\ 2.00008059515619 \\ 9.98182427225083 \\ 23.3862163639017 \end{bmatrix} \tag{15}$$

>