```
> # Construction of strictly positive polynomials in the boundary of the SOS cone with unique SOS
    decomposition.
  # See the details in Section 5 of "Strictly positive polynomials in the border of the SOS cone", by
    S. Laplagne and M. Valdettaro.
> ################################################################################
  # Load "Rational SOS" procedures
  ################################################################################
  read("rationalSOS.mpl");
  with(rationalSOS);
```

"Opening connection with Matlab"

rationalSOS := **module**( ) ... **end module**

[*cancelDenominator*, *decompositionToMatrix*, *dimSimplex*, *evalMat*, *evalSolution*, *exactSOS*, *getCoeffs*, *getDiag*, *getExtension*, *getVars*, *homogenize*, *isHomogeneous*, *linIndepRows*, *listSubsets*, *matrixToPoly*, *minorsDet*, *nonRatCoef*, *numericSolver*, *numericSolverSubmatrix*, *numericSolverSubmatrixMaxRank*, *numericSolverSubmatrixRoundBefore*, *polyToMatrix*, *polyToMatrixVars*, *primitiveMatrix*, *randomRank*, *reduceByLinearEquation*, *reduceByLinearEquationLinear*, *roundAbs*, *roundMat*, *roundMatFloat*, *roundMatToZero*, *roundToIntMatrix*, *roundVec*, *rround*, *rrounde*, *sedumiCallMaxSpectralNorm*, *sedumiCallMaxSpectralNormSDP*, *sedumiCallObjective*, *smallToZero*, *smallToZeroMatrix*, *solveSubmatrixGeneral*, *solveToZero*, *vectorTrace*, *zeroDetSRows*, *zeroDetSys*, *zeroRows*] **(1)**

```
> # Display tables of any size
  interface(rtablesize=infinity);
```

10 **(2)**

```
> ################################################################################
   ## Example 5.1
  ## Example in the border with unique solution
  ################################################################################

  # The 4 even polynomials from Reznick paper
  p1 := x*((2-1/2)*x^2-(y^2+z^2+w^2));
  p2 := y*((2-1/2)*y^2-(x^2+z^2+w^2));
  p3 := z*((2-1/2)*z^2-(x^2+y^2+w^2));
  p4 := w*((2-1/2)*w^2-(x^2+y^2+z^2));
```

$$p1 := x\left(\frac{3}{2}x^2 - w^2 - y^2 - z^2\right)$$
$$p2 := y\left(\frac{3}{2}y^2 - w^2 - x^2 - z^2\right)$$
$$p3 := z\left(\frac{3}{2}z^2 - w^2 - x^2 - y^2\right)$$
$$p4 := w\left(\frac{3}{2}w^2 - x^2 - y^2 - z^2\right) \quad \textbf{(3)}$$

```
> # f is the sum of squares of p1, ..., p4
```

```
f := p1^2 + p2^2 + p3^2 + p4^2;
f := expand(f);
```

$$f := x^2 \left(\frac{3}{2} x^2 - w^2 - y^2 - z^2\right)^2 + y^2 \left(\frac{3}{2} y^2 - w^2 - x^2 - z^2\right)^2 + z^2 \left(\frac{3}{2} z^2 - w^2 - x^2 - y^2\right)^2 + w^2 \left(\frac{3}{2} w^2 - x^2 - y^2 - z^2\right)^2$$

$$f := 6 y^2 w^2 z^2 + 6 x^2 y^2 z^2 + 6 x^2 w^2 z^2 + 6 x^2 w^2 y^2 + \frac{9}{4} x^6 + \frac{9}{4} y^6 + \frac{9}{4} z^6 + \frac{9}{4} w^6 - 2 z^4 w^2 - 2 z^2 w^4 - 2 x^4 w^2 - 2 x^4 y^2 - 2 x^4 z^2 - 2 x^2 w^4 - 2 x^2 y^4 - 2 x^2 z^4 - 2 y^4 w^2 - 2 y^4 z^2 - 2 y^2 w^4 - 2 y^2 z^4 \quad (4)$$

```
> # We use SEDUMI to compute a SOS decomposition.
  # We do not perform facial reduction, since we are interested in the
  # solutions of maximum rank.
  out := exactSOS(f, facial = "no") :
```

"Number of indeterminates: ", 126

"Calling numerical solver SEDUMI to find the values of the remaining indeterminates..."

"SEDUMI CALL - eig"

"An exact positive definite solution could not be found for the reduced problem."

"Computing Cholesky decomposition..."

"Cholesky decomposition failed. Matrix is not positive definite. We use LDLt decomposition." **(5)**

```
> # out[3] is a matrix in the spectrahedron of maximum rank.
  # We check the eigenvalues to determine the rank
  eig(out[3]);
```

(6)

$$\left[\begin{array}{c} -4.51487636644949\ 10^{-16} \\ -2.34575829216263\ 10^{-16} \\ -1.29473733035966\ 10^{-16} \\ -7.35950555191803\ 10^{-17} \\ -1.58781317001131\ 10^{-17} \\ -1.98322074127464\ 10^{-18} \\ -5.28482448521527\ 10^{-32} \\ 5.30184935009272\ 10^{-33} \\ 1.88164479185090\ 10^{-19} \\ 4.71557898349187\ 10^{-17} \\ 9.83136819637780\ 10^{-17} \\ 1.45601689705761\ 10^{-16} \\ 3.79465509016983\ 10^{-16} \\ 4.02104697381730\ 10^{-16} \\ 4.69456826983296\ 10^{-16} \\ 1.35518227929072\ 10^{-15} \\ 5.25000000000000 \\ 5.25000000000000 \\ 5.25000000000000 \\ 5.25000000000000 \end{array}\right] \tag{6}$$

```
> # We compare the matrix obtained by SEDUMI with the matrix corresponding
  # to the original decomposition p1^2+p2^2+p3^2+p4^2.
  v := convert(out[5], list) : # The monomials indexing the columns of the Gram Matrix
  A1 := decompositionToMatrix([p1, p2, p3, p4], v) :
  A2 := out[3] :
  Norm(A1 - A2);
```

$$0 \tag{7}$$

```
> # We see that both matrices are the same.
  # This gives strong numerical evidence that this is the unique matrix
  # in the spectrahedron of f.
```

```
> ###############################################################################
  ## Proposition 5.2
  ## Computational proof of the uniquenes of the SOS decomposition
  ## of Example 5.1
  ###############################################################################
```

```
> # Some preparation (computing all monomials of degree 3 and 6 in 4 variables)
  polVars := [x, y, z, w] :
  d := 3;
  varSum := add(polVars[i], i = 1 ..nops(polVars)) :
  md := expand((varSum)^d) :
  cfs := coeffs(md, polVars, 'ct3') :
  print("Monomials of degree 3: ", ct3);
  m2d := expand(varSum^(2 * d)) :
  cfs := coeffs(m2d, polVars, 'ct6') :
  print("Monomials of degree 6: ", ct6);
```

$$d := 3$$

"Monomials of degree 3: ", $w^3, w^2 x, w^2 y, w^2 z, w x^2, w x y, w x z, w y^2, w y z, w z^2, x^3, x^2 y, x^2 z, x y^2, x y z, x z^2, y^3, y^2 z, y z^2, z^3$

"Monomials of degree 6: ", $w^6, w^5 x, w^5 y, w^5 z, x^2 w^4, w^4 x y, w^4 x z, y^2 w^4, w^4 y z, z^2 w^4, w^3 x^3, w^3 x^2 y, w^3 x^2 z, w^3 x y^2, w^3 x y z, w^3 x z^2, w^3 y^3, w^3 y^2 z, w^3 y z^2, w^3 z^3, x^4 w^2, w^2 x^3 y, w^2 x^3 z, x^2 w^2 y^2, w^2 x^2 y z, x^2 w^2 z^2, w^2 x y^3, w^2 x y^2 z, w^2 x y z^2, w^2 x z^3, y^4 w^2, w^2 y^3 z, y^2 w^2 z^2, w^2 y z^3, z^4 w^2, w x^5, w x^4 y, w x^4 z, w x^3 y^2, w x^3 y z, w x^3 z^2, w x^2 y^3, w x^2 y^2 z, w x^2 y z^2, w x^2 z^3, w x y^4, w x y^3 z, w x y^2 z^2, w x y z^3, w x z^4, w y^5, w y^4 z, w y^3 z^2, w y^2 z^3, w y z^4, w z^5, x^6, x^5 y, x^5 z, x^4 y^2, x^4 y z, x^4 z^2, x^3 y^3, x^3 y^2 z, x^3 y z^2, x^3 z^3, x^2 y^4, x^2 y^3 z, x^2 y^2 z^2, x^2 y z^3, x^2 z^4, x y^5, x y^4 z, x y^3 z^2, x y^2 z^3, x y z^4, x z^5, y^6, y^5 z, y^4 z^2, y^3 z^3, y^2 z^4, y z^5, z^6$ **(8)**

```
> # We define a generic polynomial of degree d with coefficientes h_i.
  hCoeff := [h[1]] :
  for i from 2 to nops([ct3]) do
   hCoeff := [op(hCoeff), h[i]] :
  end do:
  hd := add(hCoeff[i] * ct3[i], i = 1 ..nops(hCoeff)) :
  print("Generic polynomial h of degree d: ", hd);

  hd_square := expand(hd^2) :
  aa := getCoeffs(expand(hd_square), [ct6]) :
```

"Generic polynomial h of degree d: ", $w^3 h_1 + w^2 x h_2 + w^2 y h_3 + w^2 z h_4 + w x^2 h_5 + w x y h_6 + w x z h_7 + w y^2 h_8 + w y z h_9 + w z^2 h_{10} + x^3 h_{11} + x^2 y h_{12} + x^2 z h_{13} + x y^2 h_{14} + x y z h_{15} + x z^2 h_{16} + y^3 h_{17} + y^2 z h_{18} + y z^2 h_{19} + z^3 h_{20}$ **(9)**

```
> # The 4 even polynomials from Reznick paper
  p1 := x * ((2-1/2) * x^2 - (y^2 + z^2 + w^2));
  p2 := y * ((2-1/2) * y^2 - (x^2 + z^2 + w^2));
  p3 := z * ((2-1/2) * z^2 - (x^2 + y^2 + w^2));
  p4 := w * ((2-1/2) * w^2 - (x^2 + y^2 + z^2));
```

$$p1 := x\left(\frac{3}{2}x^2 - w^2 - y^2 - z^2\right)$$

$$p2 := y\left(\frac{3}{2}y^2 - w^2 - x^2 - z^2\right)$$

$$p3 := z\left(\frac{3}{2}z^2 - w^2 - x^2 - y^2\right)$$

$$p4 := w\left(\frac{3}{2}w^2 - x^2 - y^2 - z^2\right) \tag{10}$$

```
> # In order to prove that the given decomposition is unique, we need to
  # find a PSD form whose kernel is only these 4 polynomials
```

```
> # We compute all the restrictions to phi: A6 -> R given
  # by the four polynomials. There are 20 restrictions for each polynomial
```

```
> for i from 1 to nops([ct3]) do
   p1t := expand(p1 * ct3[i]);
   vec := LinearAlgebra[Transpose](getCoeffs(p1t, [ct6]));
   if (i = 1) then
    M := ⟨vec⟩;
   else
    M := ⟨M, vec⟩;
   end if;
  end do:
  for i from 1 to nops([ct3]) do
   p2t := expand(p2 * ct3[i]);
   vec := LinearAlgebra[Transpose](getCoeffs(p2t, [ct6]));
   M := ⟨M, vec⟩;
  end do:
  for i from 1 to nops([ct3]) do
   p3t := expand(p3 * ct3[i]);
   vec := LinearAlgebra[Transpose](getCoeffs(p3t, [ct6]));
   M := ⟨M, vec⟩;
  end do:
  for i from 1 to nops([ct3]) do
   p4t := expand(p4 * ct3[i]);
   vec := LinearAlgebra[Transpose](getCoeffs(p4t, [ct6]));
   M := ⟨M, vec⟩;
  end do:
```

```
> # We solve the system using only these 80 restriction
  B := Vector(80) :
  s := LinearAlgebra[LinearSolve](M, B) :
  varss := indets(s);
  nops(varss); # 10 indeterminates left to solve
```

$$varss := \{\_t1_{24}, \_t1_{25}, \_t1_{26}, \_t1_{33}, \_t1_{41}, \_t1_{53}, \_t1_{54}, \_t1_{69}, \_t1_{74}, \_t1_{75}\}$$

$$10 \tag{11}$$

```
> # This is the expected number of indeterminates.
  # The original space has dimension 84, and the restrictions
  # 20 + 19 + 18 + 17 = 74 (because pi*pj=pj*pi give the same restriction)
```

```
> # To construct the desired form we add a new polynomial in the kernel.
  # We will find different psd forms and then add them so that
  # the kernel is generated by just the 4 polynomials
```

```
> ###############
  ## p5 := x^3;

  M2 := M:
  p5x := x^3;
  for i from 1 to nops([ct3]) do
   pst := expand(p5x * ct3[i]);
   vec := LinearAlgebra[Transpose](getCoeffs(pst, [ct6]));
   M2 := ⟨M2, vec⟩;
  end do:
  B := Vector(100) :
  s := LinearAlgebra[LinearSolve](M2, B) :
  varss := indets(s); # 1 -> We got a unique PSD form

  # We compute the form and verify it is PSD
  s1 := eval(s, {varss[1]=1}) :
  ex := LinearAlgebra[Transpose](s1) . aa :
  outx := exactSOS(ex, facial="no") :
  eig(outx[3]); # 7 positive eigenvalues and 3 null eigenvalues

  # Note that this also prooves that the sum p1^2+p2^2+p3^2+p4^2+p5^2 is
  # in the border, because we have a psd form that vanishes in this
  # five polynomials and it is not null.
```

$$p5x := x^3$$

$$varss := \{\_t4_{33}\}$$

"Number of indeterminates: ", 0

"An exact solution was found without calling the numerical solver. The solution matrix is unique under the specified conditions."

"Computing Cholesky decomposition..."

"Cholesky decomposition failed. Matrix is not positive definite. We use LDLt decomposition."

"Extension succedeed. An exact SOS decomposition has been found for the input polynomial."

(12)

$$\begin{bmatrix} 1.76469505674959\ 10^{-10} \\ 1.76470350224529\ 10^{-10} \\ 1.76470626085972\ 10^{-10} \\ 1. \\ 1. \\ 1. \\ 1. \\ 5.66666666682353 \\ 5.66666666682353 \\ 5.66666666682353 \end{bmatrix} \tag{12}$$

```
> ###############
  ## p5 := y^3;

  M2 := M:
  p5y := y^3;
  for i from 1 to nops([ct3]) do
   pst := expand(p5y * ct3[i]);
   vec := LinearAlgebra[Transpose](getCoeffs(pst, [ct6]));
   M2 := <M2, vec>;
  end do:
  B := Vector(100) :
  s := LinearAlgebra[LinearSolve](M2, B) :
  varss := indets(s); # 1 -> We got a unique PSD form

  # We compute the form and verify it is PSD
  s1 := eval(s, {varss[1] = 1}) :
  ey := LinearAlgebra[Transpose](s1) . aa :
  outy := exactSOS(ey, facial = "no") :
  eig(outy[3]); # 7 positive eigenvalues and 3 null eigenvalues
```

$$p5y := y^3$$

$$varss := \{\_t6_{26}\}$$

"Number of indeterminates: ", 0

"An exact solution was found without calling the numerical solver. The solution matrix is unique under the specified conditions."

"Computing Cholesky decomposition..."

"Cholesky decomposition failed. Matrix is not positive definite. We use LDLt decomposition."

"Extension succedeed. An exact SOS decomposition has been found for the input polynomial."

$$\begin{bmatrix} 1.76470393853378\ 10^{-10} \\ 1.76470393853378\ 10^{-10} \\ 1.76470647123006\ 10^{-10} \\ 1.00000000000000 \\ 1. \\ 1.00000000000000 \\ 1.00000000000000 \\ 5.66666666682353 \\ 5.66666666682353 \\ 5.66666666682353 \end{bmatrix} \tag{13}$$

```
> ###############
  ## p5 := z^3;

  M2 := M:
  p5z := z^3;
  for i from 1 to nops([ct3]) do
   pst := expand(p5z * ct3[i]);
   vec := LinearAlgebra[Transpose](getCoeffs(pst, [ct6]));
   M2 := <M2, vec>;
  end do:
  B := Vector(100) :
  s := LinearAlgebra[LinearSolve](M2, B) :
  varss := indets(s); # 1 -> We got a unique PSD form

  # We compute the form and verify it is PSD
  s1 := eval(s, {varss[1] = 1}) :
  ez := LinearAlgebra[Transpose](s1) . aa :
  outz := exactSOS(ez, facial = "no") :
  eig(outz[3]); # 7 positive eigenvalues and 3 null eigenvalues
```

$$p5z := z^3$$

$$varss := \{\_t8_{24}\}$$

"Number of indeterminates: ", 0

"An exact solution was found without calling the numerical solver. The solution matrix is unique under the specified conditions."

"Computing Cholesky decomposition..."

"Cholesky decomposition failed. Matrix is not positive definite. We use LDLt decomposition."

"Extension succedeed. An exact SOS decomposition has been found for the input polynomial."

$$
\left[\begin{array}{c}
1.76470393853378\ 10^{-10}\\
1.76470393853378\ 10^{-10}\\
1.76470647123006\ 10^{-10}\\
1.00000000000000\\
1.\\
1.00000000000000\\
1.00000000000000\\
5.66666666682353\\
5.66666666682353\\
5.66666666682353
\end{array}\right] \tag{14}
$$

```
> ###############
  ## p5 := w^3;

  M2 := M:
  p5w := w^3;
  for i from 1 to nops([ct3]) do
   pst := expand(p5w * ct3[i]);
   vec := LinearAlgebra[Transpose](getCoeffs(pst, [ct6]));
   M2 := <M2, vec>;
  end do:
  B := Vector(100) :
  s := LinearAlgebra[LinearSolve](M2, B) :
  varss := indets(s); # 1 -> We got a unique PSD form

  # We compute the form and verify it is PSD
  s1 := eval(s, {varss[1] = 1}) :
  ew := LinearAlgebra[Transpose](s1) . aa :
  outw := exactSOS(ew, facial = "no") :
  eig(outw[3]); # 7 positive eigenvalues and 3 null eigenvalues
```

$$p5w := w^3$$

$$varss := \{\_t10_{69}\}$$

"Number of indeterminates: ", 0

"An exact solution was found without calling the numerical solver. The solution matrix is unique under the specified conditions."

"Computing Cholesky decomposition..."

"Cholesky decomposition failed. Matrix is not positive definite. We use LDLt decomposition."

"Extension succedeed. An exact SOS decomposition has been found for the input polynomial."

$$\begin{bmatrix} 1.76470113201203\ 10^{-10} \\ 1.76470708052929\ 10^{-10} \\ 1.76471389313005\ 10^{-10} \\ 1.00000000000000 \\ 1. \\ 1.00000000000000 \\ 1.00000000000000 \\ 5.66666666682353 \\ 5.66666666682353 \\ 5.66666666682353 \end{bmatrix} \quad (15)$$

> ### The desidered form is the sum of all the rank 7 forms:
> $eall := ex + ey + ez + ew;$
> $outall := exactSOS(eall, facial = \text{"no"}) :$
> $Eigenvalues(outall[3]);$

$$eall := 8 h_1^2 + 8 h_1 h_5 + 8 h_1 h_8 + 8 h_1 h_{10} + 4 h_2^2 + 8 h_2 h_{11} + 2 h_2 h_{14} + 2 h_2 h_{16} + 4 h_3^2 + 2 h_3 h_{12} + 8 h_3 h_{17} + 2 h_3 h_{19} + 4 h_4^2 + 2 h_4 h_{13} + 2 h_4 h_{18} + 8 h_4 h_{20} + 4 h_5^2 + 2 h_5 h_8 + 2 h_5 h_{10} + h_6^2 + h_7^2 + 4 h_8^2 + 2 h_8 h_{10} + h_9^2 + 4 h_{10}^2 + 8 h_{11}^2 + 8 h_{11} h_{14} + 8 h_{11} h_{16} + 4 h_{12}^2 + 8 h_{12} h_{17} + 2 h_{12} h_{19} + 4 h_{13}^2 + 2 h_{13} h_{18} + 8 h_{13} h_{20} + 4 h_{14}^2 + 2 h_{14} h_{16} + h_{15}^2 + 4 h_{16}^2 + 8 h_{17}^2 + 8 h_{17} h_{19} + 4 h_{18}^2 + 8 h_{18} h_{20} + 4 h_{19}^2 + 8 h_{20}^2$$

"Number of indeterminates: ", 0

"An exact solution was found without calling the numerical solver. The solution matrix is unique under the specified conditions."

"Computing Cholesky decomposition..."

"Cholesky decomposition failed. Matrix is not positive definite. We use LDLt decomposition."

"Extension succedeed. An exact SOS decomposition has been found for the input polynomial."

$$\begin{bmatrix} 0 \\ 0 \\ 0 \\ 0 \\ 14 \\ 14 \\ 14 \\ 14 \\ 1 \\ 1 \\ 1 \\ 1 \\ 3 \\ 3 \\ 3 \\ 3 \\ 3 \\ 3 \\ 3 \\ 3 \end{bmatrix} \tag{16}$$

```
> # [0, 0, 0, 0, 14, 14, 14, 14, 1, 1, 1, 1, 3, 3, 3, 3, 3, 3, 3, 3]
  # (we used Eigenvalues to compute the exact values)
  # We get a psd form of rank four and p1, p2, p3, p4 are in the kernel
  # so this is form we were looking for.
```

```
> ###########################################################################
  ## Example 5.3
  ## Sum of 5 squares with unique SOS decomposition.
  ###########################################################################

  p1c := x*((2-1/2)*x^2-(y^2+z^2+w^2));
  p2c := y*((2-1/2)*y^2-(x^2+z^2+w^2));
  p3c := z*((2-1/2)*z^2-(x^2+y^2+w^2));
  p4c := w*((2-1/2)*w^2-(x^2+y^2+z^2));
  p5c := w*y*z;
  f := p1c^2+p2c^2+p3c^2+p4c^2+p5c^2;
  out := exactSOS(f, facial="no"):
  eig(out[3]);
```

$$p1c := x\left(\frac{3}{2}x^2 - w^2 - y^2 - z^2\right)$$

$$p2c := y\left(\frac{3}{2}y^2 - w^2 - x^2 - z^2\right)$$

$$p3c := z\left(\frac{3}{2}z^2 - w^2 - x^2 - y^2\right)$$

$$p4c := w\left(\frac{3}{2}w^2 - x^2 - y^2 - z^2\right)$$

$$p5c := w\,y\,z$$

$$f := x^2\left(\frac{3}{2}x^2 - w^2 - y^2 - z^2\right)^2 + y^2\left(\frac{3}{2}y^2 - w^2 - x^2 - z^2\right)^2 + z^2\left(\frac{3}{2}z^2 - w^2 - x^2 - y^2\right)^2 + w^2\left(\frac{3}{2}w^2 - x^2 - y^2 - z^2\right)^2 + y^2 w^2 z^2$$

"Number of indeterminates: ", 126

"Calling numerical solver SEDUMI to find the values of the remaining indeterminates..."

"SEDUMI CALL - eig"

"An exact positive definite solution could not be found for the reduced problem."

"Computing Cholesky decomposition..."

"Cholesky decomposition failed. Matrix is not positive definite. We use LDLt decomposition."

(17)

$$\left[\begin{array}{c} -3.86550732745581\ 10^{-16} \\ -2.43326273668495\ 10^{-16} \\ -1.44062232760108\ 10^{-16} \\ -8.43592612056251\ 10^{-17} \\ -8.15173339702456\ 10^{-17} \\ -2.66383519152873\ 10^{-17} \\ -2.56660591460691\ 10^{-17} \\ -6.07133470243070\ 10^{-19} \\ 1.49181254370184\ 10^{-17} \\ 7.76361519198023\ 10^{-17} \\ 9.57360950979169\ 10^{-17} \\ 3.25185352030438\ 10^{-16} \\ 3.80144051693241\ 10^{-16} \\ 4.49347382153596\ 10^{-16} \\ 1.34084645095915\ 10^{-15} \\ 1. \\ 5.25000000000000 \\ 5.25000000000000 \\ 5.25000000000000 \\ 5.25000000000000 \end{array}\right] \tag{17}$$

```
> # There are only 5 non-zero eigenvalues, the maximum rank in the
  # spectrahedron is 5.
```

```
> # We compare the matrix obtained by SEDUMI with the matrix corresponding
  # to the original decomposition p1c^2+p2c^2+p3c^2+p4c^2+p5c^2.
  v := convert(out[5], list) : # The monomials indexing the columns of the Gram Matrix
  A1 := decompositionToMatrix([p1c, p2c, p3c, p4c, p5c], v) :
  A2 := out[3] :
  Norm(A1 - A2);
```

$$0 \tag{18}$$

```
> ################################################################################

  ## Construction of the bilinear form to prove uniqueness of the
  ## decomposition.
```

```
# Some preparation (computing all monomials of degree 3 and 6 in 4 variables)
polVars := [x, y, z, w] :
d := 3;
varSum := add(polVars[i], i=1 ..nops(polVars)) :
md := expand((varSum)^d) :
cfs := coeffs(md, polVars, 'ct3') :
print("Monomials of degree 3: ", ct3);
m2d := expand(varSum^(2*d)) :
cfs := coeffs(m2d, polVars, 'ct6') :
print("Monomials of degree 6: ", ct6);

# We define a generic polynomial of degree d with coefficientes h_i.
hCoeff := [h[1]] :
for i from 2 to nops([ct3]) do
  hCoeff := [op(hCoeff), h[i]] :
end do:
hd := add(hCoeff[i] * ct3[i], i=1 ..nops(hCoeff) ) :
print("Generic polynomial h of degree d: ", hd);

hd_square := expand(hd^2) :
aa := getCoeffs(expand(hd_square), [ct6]) :

# The 4 even polynomials from Reznick paper and the fifth polynomial
p1 := x*((2-1/2)*x^2-(y^2 + z^2 + w^2));
p2 := y*((2-1/2)*y^2-(x^2 + z^2 + w^2));
p3 := z*((2-1/2)*z^2-(x^2 + y^2 + w^2));
p4 := w*((2-1/2)*w^2-(x^2 + y^2 + z^2));
p5 := y*z*w;
```

$$d := 3$$

"Monomials of degree 3: ", $w^3, w^2 x, w^2 y, w^2 z, w x^2, w x y, w x z, w y^2, w y z, w z^2, x^3, x^2 y, x^2 z, x y^2, x y z, x z^2, y^3, y^2 z, y z^2, z^3$

"Monomials of degree 6: ", $w^6, w^5 x, w^5 y, w^5 z, x^2 w^4, w^4 x y, w^4 x z, y^2 w^4, w^4 y z, z^2 w^4, w^3 x^3, w^3 x^2 y, w^3 x^2 z, w^3 x y^2, w^3 x y z, w^3 x z^2, w^3 y^3, w^3 y^2 z, w^3 y z^2, w^3 z^3, x^4 w^2, w^2 x^3 y, w^2 x^3 z, x^2 w^2 y^2, w^2 x^2 y z, x^2 w^2 z^2, w^2 x y^3, w^2 x y^2 z, w^2 x y z^2, w^2 x z^3, y^4 w^2, w^2 y^3 z, y^2 w^2 z^2, w^2 y z^3, z^4 w^2, w x^5, w x^4 y, w x^4 z, w x^3 y^2, w x^3 y z, w x^3 z^2, w x^2 y^3, w x^2 y^2 z, w x^2 y z^2, w x^2 z^3, w x y^4, w x y^3 z, w x y^2 z^2, w x y z^3, w x z^4, w y^5, w y^4 z, w y^3 z^2, w y^2 z^3, w y z^4, w z^5, x^6, x^5 y, x^5 z, x^4 y^2, x^4 y z, x^4 z^2, x^3 y^3, x^3 y^2 z, x^3 y z^2, x^3 z^3, x^2 y^4, x^2 y^3 z, x^2 y^2 z^2, x^2 y z^3, x^2 z^4, x y^5, x y^4 z, x y^3 z^2, x y^2 z^3, x y z^4, x z^5, y^6, y^5 z, y^4 z^2, y^3 z^3, y^2 z^4, y z^5, z^6$

"Generic polynomial h of degree d: ", $w^3 h_1 + w^2 x h_2 + w^2 y h_3 + w^2 z h_4 + w x^2 h_5 + w x y h_6 + w x z h_7 + w y^2 h_8 + w y z h_9 + w z^2 h_{10} + x^3 h_{11} + x^2 y h_{12} + x^2 z h_{13} + x y^2 h_{14} + x y z h_{15} + x z^2 h_{16} + y^3 h_{17} + y^2 z h_{18} + y z^2 h_{19} + z^3 h_{20}$

$$p1 := x\left(\frac{3}{2}x^2 - w^2 - y^2 - z^2\right)$$
$$p2 := y\left(\frac{3}{2}y^2 - w^2 - x^2 - z^2\right)$$
$$p3 := z\left(\frac{3}{2}z^2 - w^2 - x^2 - y^2\right)$$
$$p4 := w\left(\frac{3}{2}w^2 - x^2 - y^2 - z^2\right)$$
$$p5 := w\,y\,z \qquad (19)$$

```
> # In order to prove that the given decomposition is unique, we need to
  # find a PSD form whose kernel consists of only these 5 polynomials
```

```
> # We compute all the restrictions to phi: A6 -> R given
  # by the four polynomials. There are 20 restrictions for each polynomial

  for i from 1 to nops([ct3]) do
   p1t := expand(p1 * ct3[i]);
   vec := LinearAlgebra[Transpose](getCoeffs(p1t, [ct6]));
   if (i = 1) then
     M := ⟨vec⟩;
   else
     M := ⟨M, vec⟩;
   end if;
  end do:
  for i from 1 to nops([ct3]) do
   p2t := expand(p2 * ct3[i]);
   vec := LinearAlgebra[Transpose](getCoeffs(p2t, [ct6]));
   M := ⟨M, vec⟩;
  end do:
  for i from 1 to nops([ct3]) do
   p3t := expand(p3 * ct3[i]);
   vec := LinearAlgebra[Transpose](getCoeffs(p3t, [ct6]));
   M := ⟨M, vec⟩;
  end do:
  for i from 1 to nops([ct3]) do
   p4t := expand(p4 * ct3[i]);
   vec := LinearAlgebra[Transpose](getCoeffs(p4t, [ct6]));
   M := ⟨M, vec⟩;
  end do:
  for i from 1 to nops([ct3]) do
   p5t := expand(p5 * ct3[i]);
   vec := LinearAlgebra[Transpose](getCoeffs(p5t, [ct6]));
   M := ⟨M, vec⟩;
  end do:
```

```
> # We solve the system using only these 100 restrictions
  B := Vector(100) :
```

```
s := LinearAlgebra[LinearSolve](M, B) :
varss := indets(s);
nops(varss); # 3 indeterminates left to solve
```

$$varss := \{\_t14_{24}, \_t14_{26}, \_t14_{69}\}$$
$$3 \tag{20}$$

```
> # We need to assign values to these unknowns so that the resulting
  # bilinear form is positive semidefinite.
  # We try arbitrary values

  s1 := eval(s, {varss[1] = 1, varss[2] = 1, varss[3] = 1}) :
```

```
> # We compute the quadratic form
  Qform := LinearAlgebra[Transpose](s1) . aa :
  # We use exact SOS to compute the associated matrix and we check if it
  # is positive semidefinite.
  out := exactSOS(Qform, facial = "no") :
  eig(out[3]);
```

"Number of indeterminates: ", 0

"An exact solution was found without calling the numerical solver. The solution matrix is unique under the specified conditions."

"Computing Cholesky decomposition..."

"Cholesky decomposition failed. Matrix is not positive definite. We use LDLt decomposition."

"Extension succedeed. An exact SOS decomposition has been found for the input polynomial."

(21)

$$\left[\begin{array}{c} -1.42857170448147\ 10^{-10} \\ -1.42856483801333\ 10^{-10} \\ -1.42855862658907\ 10^{-10} \\ -1.45779363420685\ 10^{-16} \\ 0.999999999999999 \\ 1.00000000000000 \\ 1.00000000000000 \\ 1.41095155505225 \\ 1.41095155505225 \\ 1.41095155505225 \\ 2.00000000000000 \\ 2.00000000000000 \\ 2.00000000000000 \\ 3.00000000000000 \\ 3.00000000000000 \\ 9.92238177809060 \\ 9.92238177809060 \\ 9.92238177809061 \\ 14.0000000000000 \end{array}\right] \quad \textbf{(21)}$$

```
> # We get 15 positive eigenvalues and 4 null eigenvalues, but the
  # size of Q should be 20x20, there is one variable that does not
  # show up in Qform.
  # If we add this variable, we would get a 20x20 matrix for Q with 5
  # null-eigenvalues. This gives the desired positive semidefinite
  # quadratic form whose kernel consists of the 5 polynomials of the
  # original decomposition.

  # This implies that there cannot be more that 5 polynomials in the
  # decomposition and the decomposition as sum of 5 polynomials is
  # unique. This also implies that there cannot be decompositions with
  # less than 5 polynomials.
```

```
> ###############################################################################
  ## Example 5.4
  ## Sum of 6 squares with unique SOS decomposition.
  ###############################################################################
```

```
> p1c := x*((2-1/2)*x^2-(y^2+z^2+w^2));
  p2c := y*((2-1/2)*y^2-(x^2+z^2+w^2));
```

```
p3c := z*((2-1/2)*z^2-(x^2 + y^2 + w^2));
p4c := w*((2-1/2)*w^2-(x^2 + y^2 + z^2));
p5c := x*y*z + x*z*w;
p6c := x^2*y + x*y^2;
f := p1c^2 + p2c^2 + p3c^2 + p4c^2 + p5c^2 + p6c^2;
out := exactSOS(f, facial = "no") :
eig(out[3]);
```

$$p1c := x\left(\frac{3}{2}x^2 - w^2 - y^2 - z^2\right)$$

$$p2c := y\left(\frac{3}{2}y^2 - w^2 - x^2 - z^2\right)$$

$$p3c := z\left(\frac{3}{2}z^2 - w^2 - x^2 - y^2\right)$$

$$p4c := w\left(\frac{3}{2}w^2 - x^2 - y^2 - z^2\right)$$

$$p5c := w\,x\,z + x\,y\,z$$

$$p6c := x^2 y + x\,y^2$$

$$f := x^2\left(\frac{3}{2}x^2 - w^2 - y^2 - z^2\right)^2 + y^2\left(\frac{3}{2}y^2 - w^2 - x^2 - z^2\right)^2 + z^2\left(\frac{3}{2}z^2 - w^2 - x^2 - y^2\right)^2 + w^2\left(\frac{3}{2}w^2 - x^2 - y^2 - z^2\right)^2 + (w\,x\,z + x\,y\,z)^2 + \left(x^2 y + x\,y^2\right)^2$$

"Number of indeterminates: ", 126

"Calling numerical solver SEDUMI to find the values of the remaining indeterminates..."

"SEDUMI CALL - eig"

"An exact positive definite solution could not be found for the reduced problem."

"The computed matrix is not positive semidefinite (non-zero entries below a zero element in the diagonal). SOS decomposition may not exist."

(22)

$$\begin{bmatrix} -2.61710740488621\ 10^{-12} \\ -5.12828657062890\ 10^{-13} \\ -2.96912586583133\ 10^{-13} \\ -2.33882248240946\ 10^{-13} \\ -1.51302323438079\ 10^{-13} \\ -2.54131349181275\ 10^{-14} \\ -9.56718858672272\ 10^{-18} \\ 1.04626240822155\ 10^{-7} \\ 3.14973689686364\ 10^{-7} \\ 4.74742394648368\ 10^{-7} \\ 6.97685508756431\ 10^{-7} \\ 9.92282395541071\ 10^{-7} \\ 0.00000271621114030601 \\ 0.00000297946879896479 \\ 1.47078793119550 \\ 1.99998949061214 \\ 5.24999493459492 \\ 5.25000258435484 \\ 5.25000273945244 \\ 5.77920933357099 \end{bmatrix} \quad \textbf{(22)}$$

```
> # There are only 6 non-zero eigenvalues, the maximum rank in the
  # spectrahedron is 6.
```

```
> # We compare the matrix obtained by SEDUMI with the matrix corresponding
  # to the original decomposition p1c^2+p2c^2+p3c^2+p4c^2+p5c^2+p6c^2.
  v := convert(out[5], list) : # The monomials indexing the columns of the Gram Matrix
  A1 := decompositionToMatrix([p1c, p2c, p3c, p4c, p5c, p6c], v) :
  A2 := out[3] :
   evalf(Norm(A1 - A2));
```

$$0.00002894417271 \quad \textbf{(23)}$$

```
> # We see that both matrices are almost the same.
  # This gives strong numerical evidence that this is the unique matrix
  # in the spectrahedron of f.
```

```
>
```